

Arity-Generic Datatype-Generic Programming

Stephanie Weirich Chris Casinghino

University of Pennsylvania
{sweirich,ccasin}@cis.upenn.edu

Abstract

Some programs are doubly-generic. For example, `map` is datatype-generic in that many different data structures support a mapping operation. A generic programming language like Generic Haskell can use a single definition to generate `map` for each type. However, `map` is also arity-generic because it belongs to a family of related operations that differ in the number of arguments. For lists, this family includes `repeat`, `map`, `zipWith`, `zipWith3`, `zipWith4`, etc. With dependent types or clever programming, one can unify all of these functions together in a single definition.

However, no one has explored the combination of these two forms of genericity. These two axes are not orthogonal because the idea of arity appears in Generic Haskell: datatype-generic versions of `repeat`, `map` and `zipWith` have different arities of kind-indexed types. In this paper, we define arity-generic datatype-generic programs by building a framework for Generic Haskell-style generic programming in the dependently-typed programming language Agda 2.

Categories and Subject Descriptors D.1.1 [*Programming Techniques*]: Applicative (Functional) Programming; D.3.3 [*Program-*

ming languages]: Language constructs and features—Data types and structures, frameworks

General Terms Design, Languages, Verification

Keywords Dependent types, Arity-generic programming, Agda, Generic Haskell

1. Introduction

This is a story about doubly-generic programming. *Datatype-generic* programming defines operations that may be instantiated at many different types, so these operations need not be redefined for each one. For example, Generic Haskell [5, 11] includes a generic map operation `gmap` that has instances for types such as lists, optional values, and products (even though these types have different kinds).

$$\begin{aligned} \text{gmap } \langle [] \rangle &:: (a \rightarrow b) \rightarrow [a] \rightarrow [b] \\ \text{gmap } \langle \text{Maybe} \rangle &:: (a \rightarrow b) \rightarrow \text{Maybe } a \rightarrow \text{Maybe } b \\ \text{gmap } \langle (,) \rangle &:: (a1 \rightarrow b1) \rightarrow (a2 \rightarrow b2) \\ &\rightarrow (a1, a2) \rightarrow (b1, b2) \end{aligned}$$

Permission to make digital or hard copies of all or part of this work for personal or classroom use is granted without fee provided that copies are not made or distributed for profit or commercial advantage and that copies bear this notice and the full citation on the first page. To copy otherwise, to republish, to post on servers or to redistribute to lists, requires prior specific permission and/or a fee.

Because all the instances of `gmap` are generated from the same definition, reasoning about that generic function tells us about `map` at each type.

However, there is another way to generalize `map`. Consider the following sequence of functions from the Haskell Prelude [20], all of which operate on lists.

```
repeat    :: a → [a]
map       :: (a → b) → [a] → [b]
zipWith   :: (a → b → c) → [a] → [b] → [c]
zipWith3  :: (a → b → c → d) → [a] → [b] → [c] → [d]
```

The `repeat` function creates an infinite list from its argument. The `zipWith` function is a generalization of `zip`—it combines the two lists together with its argument instead of with the tupling function. Likewise, `zipWith3` combines three lists.

As Fridlender and Indrika [7] have pointed out, all of these functions are instances of the same generic operation, they just have different *arities*. They demonstrate how to encode the arity as a Church numeral in Haskell and uniformly produce all of these list operations from the same definition.

Arity-genericity is not unique to the list instance of `map`. It is not difficult to imagine arity-generic versions of `map` for other types. Fridlender and Indrika's technique immediately applies to all types that are applicative functors [16]. However, one may also define arity-generic versions of `map` at other types.

Other functions besides `map` have both datatype-generic and arity-generic versions. For example, equality can be applied to

any number of arguments, all of the same type. Map has a dual operation called `unzipWith` that is similarly doubly-generic. Other examples include folds, enumerations, monadic maps, etc.

In this paper, we present the first *doubly-generic* definitions. For each of these examples, we can give a single definition that can be instantiated at any type or at any arity. Our methodology shows how these two forms of genericity can be combined in the same framework and demonstrates the synergy between them.

In fact, arity-genericity is not independent of datatype-genericity. Generic Haskell has its own notion of arity, and each datatype-generic function must be defined at a particular arity. Importantly, that arity corresponds to the arities in `map` above. For example, the Generic Haskell version of `repeat` has arity one, its `map` has arity two, and `zipWith` has arity three.

Unfortunately, Generic Haskell does not permit generalizing over arities, so a single definition cannot produce `repeat`, `map` and `zipWith`. Likewise, Generic-Haskell-style libraries encoded in Haskell, such as `RepLib` [29] or `Extensible and Modular Generics for the Masses (EMGM)` [6] specialize their infrastructure to specific arities, so they too cannot write arity-generic code.

However, Altenkirch and McBride [1] and Verbruggen et al. [26] have shown how to encode Generic-Haskell-style generic programming in dependently-typed programming languages. Although they do not consider arity-genericity in their work, because of the power of dependent-types, their encodings are flexible enough to express arity-generic operations.

In this paper, we develop an analogous generic programming framework in the dependently-typed language `Agda 2` [19] and demonstrate how it may be used to define doubly-generic operations. We choose `Agda` because it is particularly tailored to

dependently-typed programming, but we could have also used a number of different languages, such as Coq [25], Epigram [15], Ω mega [23], or Haskell with recent extensions [3, 21].

Our contributions are as follows:

1. We develop an arity-generic version of `map` that reveals commonality between `gmap`, `gzipWith` and `gzipWith3`. This correspondence has not previously been expressed, but we find that it leads to insight into the nature of these operations. Since the definitions are all instances of the same dependently-typed function, we have shown formally that they are related.
2. This example is developed on top of a reusable framework for generic programming in Agda. Although our framework has the same structure as previous work, our treatment of datatype isomorphisms is novel and requires less boilerplate.
3. We use our framework to develop other doubly-generic operations, such as `equality` and `unzipWith`. All of these examples shed light on arity support in a generic programming framework. In particular, there are not many operations that require arity of two or more: this work suggests what such operations must look like.
4. Finally, because we develop a reusable framework, this work demonstrates how a tool like Generic Haskell could be extended to arity-genericity.

We explain doubly-generic `map` and our generic programming infrastructure in stages. In Section 2 we start with an Agda definition of arity-generic `map`. Next, in Section 3, we describe a general framework for generic programming that works for all types

(of any kind) formed from unit, pairs, sums and natural numbers. We use this framework to define doubly-generic map. In Section 4 we show how datatype isomorphisms may be incorporated, so that we can specialize doubly-generic operations to inductive datatypes. We discuss other doubly-generic examples in Section 5. Finally, Sections 6 and 7 discuss related work and conclude.

All code described in this paper is available from <http://www.cis.upenn.edu/~ccasin/papers/aritygen.tar.gz>.

2. Arity-generic Map

We begin this section by introducing Agda and using it to define applicative functors. We show how to use applicative functors to define arity-generic map for vectors, following Fridlender and Indrika. Finally, we demonstrate why this approach does not scale to implementing datatype-generic arity-generic map.

2.1 Programming with Dependent Types in Agda

Agda is a dependently typed programming language where terms may appear in types. For example, the Agda standard library defines a type of polymorphic length-indexed vectors:

```
data Vec (A : Set) : ℕ → Set where  
  []      : Vec A zero  
  _ :: _  : ∀ {n} (x : A) (xs : Vec A n) → Vec A (suc n)
```

This datatype `Vec` is parameterized by an argument `A` of type `Set`, the analogue of Haskell’s kind `*`, and indexed by an argument of type \mathbb{N}^1 , the type of natural numbers. The parameter `A` specifies

¹ Note that Unicode symbols are valid in Agda identifiers.

the type of the objects stored in the vector and the index specifies its length. For example, the type `Vec Bool 2` is a list of boolean values of length two. Note that indices can vary in the types of the constructors; for example, empty vectors `[]` use index 0.

The underscores in `_ :: _` create an infix operator. Arguments to Agda functions may be made implicit by placing them in curly braces, so Agda will attempt to infer the length index by unification when applying `_ :: _`. For example, Agda can automatically determine that the term `true :: false :: []` has type `Vec Bool 2`.

Vectors are applicative functors, familiar to Haskell programmers from the `Applicative` type class. Applicative functors have two operations. The first is `repeat` (called `pure` in Haskell). Given an initial value, it constructs a vector with `n` copies of that value.

$$\begin{aligned} \text{repeat} &: \{n : \mathbb{N}\} \rightarrow \{A : \text{Set}\} \rightarrow A \rightarrow \text{Vec } A \ n \\ \text{repeat } \{\text{zero}\} \ x &= [] \\ \text{repeat } \{\text{suc } n\} \ x &= x :: \text{repeat } \{n\} \ x \end{aligned}$$

Observe that, using curly braces, implicit arguments can be explicitly provided in a function call or matched against in a definition.

The second, `_ ⊗ _`, is an infix zipping application, pronounced "zap" and defined by:

$$\begin{aligned} _ \otimes _ &: \{A \ B : \text{Set}\} \{n : \mathbb{N}\} \\ &\rightarrow \text{Vec } (A \rightarrow B) \ n \rightarrow \text{Vec } A \ n \rightarrow \text{Vec } B \ n \\ [] &\otimes [] = [] \\ (a :: As) \otimes (b :: Bs) &= (a \ b :: As \otimes Bs) \end{aligned}$$

The `_ ⊗ _` operator associates to the left. In its definition, we do not need to consider the case where one vector is empty while the other is not because the type specifies that both arguments have the

same length.

These two operations are the key to arity-generic map. The following sequence shows that the different arities of map follow a specific pattern.

```
map0      : { m : ℕ } { A : Set } → A → Vec A m
map0      = repeat

map1      : { m : ℕ } { A B : Set }
           → (A → B) → Vec A m → Vec B m
map1 f x   = repeat f ⊗ x

map2      : { m : ℕ } { A B C : Set }
           → (A → B → C)
           → Vec A m → Vec B m → Vec C m
map2 f x1 x2 = repeat f ⊗ x1 ⊗ x2
```

Indeed, all of these maps are defined by a simple application of `repeat` and `n` copies of `_⊗_`. Agda can express the arity-generic operation that unifies all of these maps via dependent types, as we present in the next subsection.

2.2 Arity-Generic Vector Map

The difficulty in the definition of arity-generic map is that all of the instances have different types. Given some arity `n`, we must generate the corresponding type in this sequence. Fridlender and Indrika, not working in a dependently typed language, do so by encoding `n` as a Church numeral that generates the appropriate type for `map`.

We prefer to use natural numbers to express the arity of the mapping operation. Therefore, we must *program* with Agda types. For

example, we can construct a vector of Agda types, `Bool :: ℕ :: []`, which has type `Vec Set 2`, and use standard vector operations (such as `_*`) with this value.²

²This type requires `Set` to have type `Set`, enabled by Agda's `--type-in-type` flag. The standard type system of Agda has an infinite hierarchy of `Sets`, and users must resolve their code to be at the appropriate

The first step is to define `arrTy`, which folds the arrow type constructor `→` over a non-empty vector of types. Given such a vector, this operation constructs the type of the function that will be mapped over the `n` data structures.

$$\begin{aligned}\text{arrTy} &: \{n : \mathbb{N}\} \rightarrow \text{Vec Set (suc } n) \rightarrow \text{Set} \\ \text{arrTy } \{0\} \quad (A :: []) &= A \\ \text{arrTy } \{\text{suc } n\} \quad (A :: As) &= A \rightarrow \text{arrTy } As\end{aligned}$$

The function `arrTyVec` constructs the result type of arity-generic map for vectors. We define this operation by mapping the `Vec` constructor onto the vector of types, then placing arrows between them. Notice that there are two integer indices here: `n` determines the number of types we are dealing with (the arity), while `m` is the length of the vectors we map over. Recall that the curly braces in the types of `arrTyVec` and `arrTy` mark `m` and `n` as implicit arguments, so we need not always match against them in definitions nor provide them explicitly as arguments.

$$\begin{aligned}\text{arrTyVec} &: \{m\ n : \mathbb{N}\} \rightarrow \text{Vec Set (suc } n) \rightarrow \text{Set} \\ \text{arrTyVec } \{m\} \quad As &= \\ &\quad \text{arrTy (repeat } (\lambda A \rightarrow \text{Vec } A\ m) \text{ } *) \text{ } As)\end{aligned}$$

For example, we can define the sequence of types from `Sec-`

tion 2.1 using these functions applied to lists of type variables.

```
map0 : {m : ℕ} {A : Set}
      → arrTy (A :: [])
      → arrTyVec {m} (A :: [])
map1 : {m : ℕ} {A B : Set}
      → arrTy (A :: B :: [])
      → arrTyVec {m} (A :: B :: [])
map2 : {m : ℕ} {A B C : Set}
      → arrTy (A :: B :: C :: [])
      → arrTyVec {m} (A :: B :: C :: [])
```

Now, to define arity-generic map, we start by defining a function `nvec-map`. The type of this function mirrors the examples above, except that it takes in the type arguments (A, B, etc) as a vector (As). After we define `nvec-map` we will curry it to get the desired operation.

```
nvec-map : {m : ℕ} → (n : ℕ)
          → (As : Vec Set (suc n))
          → arrTy As → arrTyVec {m} As
```

Intuitively, the definition of `nvec-map` is a simple application of `repeat` and `n` copies of `_ * _`:

```
nvec-map As f v1 v2 ... vn =
  repeat f * v1 * v2 * ... * vn
```

We define this function by recursion on `n`, in accumulator style. After duplicating `f` we have a vector of functions to zap, so we

define a helper function, g , for that more general case.

$$\begin{aligned}
& \text{nvec-map } n \text{ As } f = g \{n\} \text{ As (repeat } f) \textbf{ where} \\
& \quad g : \{n \ m : \mathbb{N}\} \\
& \quad \rightarrow (\text{As} : \text{Vec Set (suc } n)) \\
& \quad \rightarrow \text{Vec (arrTy As)} \ m \rightarrow \text{arrTyVec } \{m\} \text{ As} \\
& \quad g \{0\} \quad (A :: []) \quad a = a \\
& \quad g \{\text{suc } n\} (A1 :: \text{As}) f = \\
& \quad \quad (\lambda a \rightarrow g \text{ As } (f \circledast a))
\end{aligned}$$

level. Although we have done so, in the interest of clarity we have hidden this hierarchy and its associated complexities. We discuss this choice further in Section 7.

Finally, we define two operations for currying. The first, $\forall \Rightarrow$, creates a curried version of a type which depends on a vector. The second, $\lambda \Rightarrow$, curries a corresponding function term.

$$\begin{aligned}
& \forall \Rightarrow : \{n : \mathbb{N}\} \{A : \text{Set}\} \rightarrow (\text{Vec } A \ n \rightarrow \text{Set}) \rightarrow \text{Set} \\
& \forall \Rightarrow \{\text{zero}\} \quad B = B [] \\
& \forall \Rightarrow \{\text{suc } n\} \{A\} \ B = \\
& \quad \{a : A\} \rightarrow \forall \Rightarrow \{n\} (\lambda as \rightarrow B (a :: as)) \\
& \lambda \Rightarrow : \{n : \mathbb{N}\} \{A : \text{Set}\} \{B : \text{Vec } A \ n \rightarrow \text{Set}\} \\
& \quad \rightarrow ((X : \text{Vec } A \ n) \rightarrow B \ X) \rightarrow (\forall \Rightarrow B) \\
& \lambda \Rightarrow \{\text{zero}\} \quad f = f [] \\
& \lambda \Rightarrow \{\text{suc } n\} \{A\} \ f = \\
& \quad \lambda \{a : A\} \rightarrow \lambda \Rightarrow \{n\} (\lambda as \rightarrow f (a :: as))
\end{aligned}$$

With these operations, we can finish the definition of arity-generic map. Again, the (implicit) argument m is the length of the

vector, and the (explicit) argument n is the specific arity of `map` that is desired.

$$\begin{aligned} \text{nmap} &: (n : \mathbb{N}) \rightarrow \{m : \mathbb{N}\} \\ &\rightarrow \forall \Rightarrow (\lambda (As : \text{Vec Set } (\text{suc } n)) \\ &\rightarrow \text{arrTy } As \rightarrow \text{arrTyVec } \{m\} As) \\ \text{nmap } n \{m\} &= \lambda \Rightarrow (\text{nvec-map } \{m\} n) \end{aligned}$$

We can use this arity-generic map just by providing the arity as an additional argument. For example, the term `nmap 1` has type

$$\begin{aligned} \{m : \mathbb{N}\} &\rightarrow \{A B : \text{Set}\} \rightarrow (A \rightarrow B) \\ &\rightarrow (\text{Vec } A \ m) \rightarrow (\text{Vec } B \ m) \end{aligned}$$

and the expression

$$\text{nmap } 1 (\lambda x \rightarrow x + 1) (1 :: 2 :: 3 :: [])$$

evaluates to `2 :: 3 :: 4 :: []`. Likewise, the term `nmap 2` has type

$$\begin{aligned} \{m : \mathbb{N}\} &\rightarrow \{A B C : \text{Set}\} \rightarrow (A \rightarrow B \rightarrow C) \\ &\rightarrow (\text{Vec } A \ m) \rightarrow (\text{Vec } B \ m) \rightarrow (\text{Vec } C \ m) \end{aligned}$$

and the expression

$$\text{nmap } 2 (_, _) (1 :: 2 :: 3 :: []) (4 :: 5 :: 6 :: [])$$

returns `(1, 4) :: (2, 5) :: (3, 6) :: []`.

2.3 Towards Type Genericity

We have shown how to define arity-generic map for vectors, but what about for other types of data, such as products of vectors or vectors of products? This should be possible, as `map` is a *type-generic* operation, one that is defined by type structure.

Type-generic programming in Agda is done via a technique called *universes* [13, 18]. The idea is to define an inductive datatype `Tyc`, called a universe, which represents types, along with an interpretation function $\llbracket _ \rrbracket$ that maps elements of this universe to actual Agda types. A generic program is then an operation which manipulates this data structure.

However, there is one difficulty—what kind of types should we represent? The answer to that question determines the type of the interpretation function. For example, if the datatype `Tyc` represents types of kind `Set` then the interpretation function should have type $\text{Tyc} \rightarrow \text{Set}$. If the universe represents type constructors, that is, functions from types to types, then the interpretation function should have type $\text{Tyc} \rightarrow (\text{Set} \rightarrow \text{Set})$.

Consider the following universe of codes for type constructors.

```
data Tyc : Set where
  Nat   : Tyc
  Unit  : Tyc
  Prod  : Tyc → Tyc → Tyc
  Arr   : ℕ → Tyc → Tyc
  Var   : Tyc
```

Each of these codes can be decoded as an Agda type constructor of kind $\text{Set} \rightarrow \text{Set}$. For example, \top is the unit type in Agda and \times constructs the (non-dependent) type of products.

```
 $\llbracket \_ \rrbracket$            : Tyc → (Set → Set)
 $\llbracket \text{Nat} \rrbracket$       a = ℕ
 $\llbracket \text{Unit} \rrbracket$      a =  $\top$ 
 $\llbracket \text{Prod } t1 \ t2 \rrbracket$  a =  $\llbracket t1 \rrbracket a \times \llbracket t2 \rrbracket a$ 
 $\llbracket \text{Arr } n \ t1 \rrbracket$  a = Vec ( $\llbracket t1 \rrbracket a$ ) n
```

$\lfloor \text{Var} \rfloor \quad a = a$

With these two definitions, we can implement type-generic versions of the `repeat` and `_*_` functions.³ They are implemented by recursion on the structure of the `Tyc`, but we elide the definitions for brevity.

$\text{repeat} : (t : \text{Tyc}) \rightarrow \{a : \text{Set}\} \rightarrow a \rightarrow \lfloor t \rfloor a$
 $\text{gzap} : (t : \text{Tyc}) \rightarrow \{a\ b : \text{Set}\}$
 $\rightarrow \lfloor t \rfloor (a \rightarrow b) \rightarrow \lfloor t \rfloor a \rightarrow \lfloor t \rfloor b$

With these type-generic functions, we can generalize the definition for vectors to produce `gmap`, which works for all type constructors in the universe. This definition is a straightforward extension of `nmap` for vectors (replacing `repeat` and `_*_` with `grepeat` and `gzap`), so we elide its definition and only show its type.

$\text{gmap} : (t : \text{Tyc}) \rightarrow (n : \mathbb{N})$
 $\rightarrow \forall \Rightarrow (\lambda (As : \text{Vec Set (suc } n))$
 $\rightarrow \text{arrTy } As \rightarrow \text{arrTy (repeat } \lfloor t \rfloor \text{ } \ast As))$

We use `gmap` by supplying a type code and arity. For example,

$\text{example-map} : \{m : \mathbb{N}\} \rightarrow \{A\ B : \text{Set}\} \rightarrow (A \rightarrow B)$
 $\rightarrow \text{Vec } (A \times (A \times \top))\ m$
 $\rightarrow \text{Vec } (B \times (B \times \top))\ m$
 $\text{example-map} =$
 $\text{gmap (Arr } _ \text{ (Prod Var (Prod Var Unit))) } 1$

We have now combined arity genericity and type genericity. However, there is a problem with this definition; it only works for type constructors of kind `Set → Set`. Maps for other kinds are not avail-

able. Furthermore, this definition tells us nothing about how to define other arity-generic functions. We have not really gotten to the essence of arity genericity.

To extend arity-generic map to types of arbitrary kinds, we will redo our framework for type-generic programming using a kind-indexed universe. The kind determines the type of the decoding function `[_]`. With this kind-indexed universe, the concept of arity naturally shows up—following Generic Haskell, a generic function has a kind-indexed type of a particular arity. For example, generic repeat requires an arity one kind-indexed type, while generic map requires arity two, and generic zipWith requires arity three. Remarkably, but perhaps unsurprisingly, this notion of arity mirrors the arity found in arity-generic map!

What is new in this paper is that we generalize over the arities in the kind-indexed types to give a completely new definition of arity-generic type-generic map. This definition incorporates arity-genericity right from the start. In the current section we layered arity-genericity on top of type-genericity; in the next, our type-generic functions will be inherently arity-generic.

3. Arity-Generic Type-Generic Map

Next, we show how to generalize arity-generic map to arbitrary type constructors by implementing a framework for Generic

³ Because these generic functions show that all type constructors in this universe are applicative functors, we cannot include a code for sum types. We return to this issue in Section 3.3.

Haskell style kind-indexed types. We develop our framework in stages, first including only primitive type constructors in the

universe, then in Section 4, extending it to include user-defined datatypes.

3.1 Universe Definition

To write more general generic programs, we need a more expressive universe. The universe that we care about is based on the type language of F-omega [8]. It is the simply-typed lambda calculus augmented with a number of constants that form types. Therefore, to represent this language, we need datatypes for kinds, constants, and for the lambda calculus itself.

Kinds include the base kind \star and function kinds. The function kind arrow associates to the right.

```
data Kind : Set where  
   $\star$       : Kind  
   $\_ \Rightarrow \_$  : Kind  $\rightarrow$  Kind  $\rightarrow$  Kind
```

A simple recursive function takes a member of this datatype into an Agda kind.

```
 $\llbracket \_ \rrbracket$       : Kind  $\rightarrow$  Set  
 $\llbracket \star \rrbracket$     = Set  
 $\llbracket a \Rightarrow b \rrbracket$  =  $\llbracket a \rrbracket \rightarrow \llbracket b \rrbracket$ 
```

Constants are indexed by their kinds. For now, we will concentrate on types formed from natural numbers, unit, binary sums, and binary products. Note that these definitions include a code for sum types. Although doubly-generic map is partial for sums, many doubly-generic operations are not. On the other hand, most generic functions are partial for function types, so we do not include a code for them. Furthermore, because vectors are representable in terms

of the other constructors, we do not include a code for them in this universe. This keeps the definitions of arity-generic functions simple. In Section 4, we discuss how our generic programming framework can interface directly with Agda datatypes like `Vec`.

```
data Const : Kind → Set where
```

```
  Nat  : Const ★
```

```
  Unit : Const ★
```

```
  Sum  : Const (★ ⇒ ★ ⇒ ★)
```

```
  Prod : Const (★ ⇒ ★ ⇒ ★)
```

Again, each of these constants can be decoded as an Agda type constructor.

```
interp-c : ∀ {k} → Const k →  $\llbracket k \rrbracket$ 
```

```
interp-c Unit =  $\top$ 
```

```
interp-c Nat  =  $\mathbb{N}$ 
```

```
interp-c Sum  =  $\_ \sqcup \_$ 
```

```
interp-c Prod =  $\_ \times \_$ 
```

To represent other types (of arbitrary kinds), we now define an indexed datatype called `Typ`. A `Typ` may be a variable, a lambda, an application, or a constant. The datatype is indexed by the kind of the type and a typing context which indicates the kinds of variables. We use de Bruijn indices for variables, so we represent the typing context as a list of `Kinds`. The n th `Kind` in the list is the kind of variable n .

```
Ctx : Set
```

```
Ctx = List Kind
```

```
data TyVar : Ctx → Kind → Set where
```

```

VZ  :  $\forall \{G\ k\} \rightarrow \text{TyVar } (k :: G) \ k$ 
VS  :  $\forall \{G\ k'\ k\} \rightarrow \text{TyVar } G\ k \rightarrow \text{TyVar } (k' :: G) \ k$ 

data Typ : Ctx  $\rightarrow$  Kind  $\rightarrow$  Set where
  Var  :  $\forall \{G\ k\} \rightarrow \text{TyVar } G\ k \rightarrow \text{Typ } G\ k$ 
  Lam  :  $\forall \{G\ k_1\ k_2\} \rightarrow \text{Typ } (k_1 :: G) \ k_2$ 
         $\rightarrow \text{Typ } G\ (k_1 \Rightarrow k_2)$ 
  App  :  $\forall \{G\ k_1\ k_2\} \rightarrow \text{Typ } G\ (k_1 \Rightarrow k_2) \rightarrow \text{Typ } G\ k_1$ 
         $\rightarrow \text{Typ } G\ k_2$ 
  Con  :  $\forall \{G\ k\} \rightarrow \text{Const } k \rightarrow \text{Typ } G\ k$ 

```

We use the notation `Ty` for closed types—those that can be checked in the empty typing context.

```

Ty : Kind  $\rightarrow$  Set
Ty = Typ []

```

Now that we can represent type constructors, we need a mechanism to decode them as Agda types. To do so, we must have an environment to decode the variables. We index the datatype for the environment with the typing context to make sure that each variable is mapped to an Agda type of the right kind. Note that this definition overloads the `[]` and `_::_` constructors, but Agda can infer which we mean.

```

data Env : List Kind  $\rightarrow$  Set where
  []      : Env []
  _::_    :  $\forall \{k\ G\} \rightarrow \llbracket k \rrbracket \rightarrow \text{Env } G \rightarrow \text{Env } (k :: G)$ 

  sLookup :  $\forall \{k\ G\} \rightarrow \text{TyVar } G\ k \rightarrow \text{Env } G \rightarrow \llbracket k \rrbracket$ 
  sLookup VZ      (v :: G) = v
  sLookup (VS x) (v :: G) = sLookup x G

```

Finally, with the help of the environment, we can decode a `Typ` as an Agda type of the appropriate kind. We use the `[_]` notation for decoding closed types in the empty environment.

```

interp : ∀ {k G} → Typ G k → Env G → [ k ]
interp (Var x) e      = sLookup x e
interp (Lam t) e      = λ y → interp t (y :: e)
interp (App t1 t2) e  = (interp t1 e) (interp t2 e)
interp (Con c) e      = interp-c c

[ _ ] : ∀ {k} → Ty k → [ k ]
[ t ] = interp t []

```

For example, the following type constructor `Option` (isomorphic to the standard `Maybe` datatype)

```

Option : Set → Set
Option = λ A → ⊤ ⊔ A

```

is represented with the following code:

```

option : Ty (★ ⇒ ★)
option =
  Lam (App (App (Con Sum) (Con Unit)) (Var VZ))

```

The Agda type checker can see that `[option]` normalizes to `Option`, so it considers these two expressions equal.

3.2 Framework for Doubly-Generic Programming

Next, we give the signature of a framework for defining arity-generic type-generic programs. For space reasons, we do not give the implementation of this framework here. The interested reader

may consult Altenkirch and McBride [1], Verbruggen et al. [26], or our source code for more details.

As with Generic Haskell, the behavior of a generic program defined using this framework is fixed for applications, lambdas and variables. Therefore, to define an arity-generic type-generic operation, we need only supply the behavior of the generic program for the type constants.

Datatype-generic operations have different types when instantiated at different kinds, so they are described by *kind-indexed types* [11]. For example, consider the type of the standard map function for the Option type constructor, of kind $\star \Rightarrow \star$:

$$\begin{aligned} \text{option-map1} &: \forall \{A\ B\} \rightarrow (A \rightarrow B) \\ &\rightarrow (\text{Option } A \rightarrow \text{Option } B) \end{aligned}$$

And map for the type constructor $_ \times _$, of kind $\star \Rightarrow \star \Rightarrow \star$

$$\begin{aligned} \text{pair-map1} &: \forall \{A1\ A2\ B1\ B2\} \\ &\rightarrow (A1 \rightarrow B1) \rightarrow (A2 \rightarrow B2) \\ &\rightarrow (A1 \times A2) \rightarrow (B1 \times B2) \end{aligned}$$

Though different, the types of `option-map1` and `pair-map1` are instances of the same kind-indexed type. In Generic Haskell, kind-indexed types are defined by recursion on the kind of the type arguments. For example, here is the Generic Haskell definition of `map`'s type [12]:

$$\begin{aligned} \text{type Map } \langle \star \rangle \ t_1 \ t_2 &= t_1 \rightarrow t_2 \\ \text{type Map } \langle k_1 \Rightarrow k_2 \rangle \ t_1 \ t_2 &= \\ \forall a_1 \ a_2, \text{Map } \langle k_1 \rangle \ a_1 \ a_2 &\rightarrow \text{Map } \langle k_2 \rangle \ (t_1 \ a_1) \ (t_2 \ a_2) \end{aligned}$$

Readers new to Generic Haskell-style generic programming may

find it instructive to verify that $\text{Map } \langle \star \Rightarrow \star \rangle$ Option Option and $\text{Map } \langle \star \Rightarrow \star \Rightarrow \star \rangle$ $_ \times _ _ \times _$ simplify to the types given above for `option-map` and `pair-map` (modulo notational differences).

For arity-genericity, we must generalize kind-indexed types in another way. We want not only `pair-map1`, but also `pair-map` at other arities to be instances as well:

`pair-map0` : $\forall \{A\ B : \text{Set}\} \rightarrow A \rightarrow B \rightarrow A \times B$

`pair-map2` : $\forall \{A1\ B1\ C1\ A2\ B2\ C2\}$
 $\rightarrow (A1 \rightarrow B1 \rightarrow C1) \rightarrow (A2 \rightarrow B2 \rightarrow C2)$
 $\rightarrow A1 \times A2 \rightarrow B1 \times B2 \rightarrow C1 \times C2$

We compute the type of a generic function instance from four pieces of information: the arity of the operation (given with an implicit argument `n`), a function `b` to construct the type in the base case, the kind `k` itself and a vector `v` of `n` Agda types, each of kind `k`. Reminiscent of Generic Haskell, our kind-indexed type is written `b < k > v`:

`_<_>_` : $\{n : \mathbb{N}\}$
 $\rightarrow (b : \text{Vec Set (suc } n) \rightarrow \text{Set})$
 $\rightarrow (k : \text{Kind})$
 $\rightarrow \text{Vec } \llbracket k \rrbracket (\text{suc } n)$
 $\rightarrow \text{Set}$

`b < ⋆ > Vs` = `b Vs`
`b < k1 ⇒ k2 > Vs` = $\forall \Rightarrow \lambda (As : \text{Vec } \llbracket k1 \rrbracket _) \rightarrow$
 $b \langle k1 \rangle As \rightarrow b \langle k2 \rangle (Vs \circledast As)$

The primary difference between our definition and the Generic

Haskell definition of kind-indexed type is that because the arity is a parameter, we deal with the type arguments as a vector rather than as individuals. For higher kinds the polymorphic type produced takes n arguments of kind $\llbracket k1 \rrbracket$ (the vector As) and a kind-indexed type for those arguments and produces a result where each higher kinded type in the vector Vs has been applied to each argument in vector As .

We use the $\forall \Rightarrow$ function (from Section 2) to curry the type so that the user may provide n individual $\llbracket k1 \rrbracket$'s rather than a vector. The $_$ in `Vec $\llbracket k1 \rrbracket$ $_$` instructs Agda to infer the length of the vector (convenient since we did not give a name to the arity).

We do not allow these vectors to be empty because few generic functions make sense at arity zero. If we had allowed empty vectors we would have to add a degenerate zero case for the majority of generic functions. It would be straightforward, but tedious, to remove this restriction. As a result, the number provided here as an arity (n) is one less than the corresponding Generic Haskell arity. We refer to this reduced number as the arity for convenience.

We define generic functions with `ngen`, whose type is shown below. This operation produces a value of a kind-indexed type given `ce`, a mapping from constants to appropriate definitions.

$$\begin{aligned} \text{ngen} &: \{n : \mathbb{N}\} \{b : \text{Vec Set (suc } n) \rightarrow \text{Set}\} \{k : \text{Kind}\} \\ &\rightarrow (t : \text{Ty } k) \\ &\rightarrow (ce : \text{TyConstEnv } n \ b) \\ &\rightarrow b \langle k \rangle (\text{repeat } \lfloor t \rfloor) \end{aligned}$$

The type of `ce` is a function which maps each constant to a value of the kind-indexed type associated with that constant.

$$\text{TyConstEnv} : \{n : \mathbb{N}\} \rightarrow (b : \text{Vec Set } n \rightarrow \text{Set}) \rightarrow \text{Set}$$

$$\text{TyConstEnv } b = \{k : \text{Kind}\} (c : \text{Const } k) \rightarrow b \langle k \rangle (\text{repeat } \lfloor \text{Con } c \rfloor)$$

We can already use this framework for non-arity-generic programming. For example, suppose we wished to define the standard generic map. In this case, we would provide the following definition for b .

$$\begin{aligned} \text{Map} &: \text{Vec Set } 2 \rightarrow \text{Set} \\ \text{Map } (A :: B :: []) &= A \rightarrow B \end{aligned}$$

Next, we define the type-constant environment for this particular b . The mapping function for natural numbers and unit is an identity function. For products and sums, the mapping function takes those arguments apart, maps the subcomponents and then puts them back together.

$$\begin{aligned} \text{gmap-const} &: \text{TyConstEnv GMap} \\ \text{gmap-const Nat} &= \lambda x \rightarrow x \\ \text{gmap-const Unit} &= \lambda x \rightarrow x \\ \text{gmap-const Prod} &= \lambda f \, g \, x \rightarrow (f (\text{proj}_1 \, x), g (\text{proj}_2 \, x)) \\ \text{gmap-const Sum} &= g \end{aligned}$$

where

$$\begin{aligned} g &: \{A1 \, B1 \, A2 \, B2 : \text{Set}\} \\ &\rightarrow (A1 \rightarrow B1) \rightarrow (A2 \rightarrow B2) \\ &\rightarrow A1 \uplus A2 \rightarrow B1 \uplus B2 \\ g \, fa \, fb \, (\text{inj}_1 \, xa) &= \text{inj}_1 \, (fa \, xa) \\ g \, fa \, fb \, (\text{inj}_2 \, xb) &= \text{inj}_2 \, (fb \, xb) \end{aligned}$$

Generic map then calls ngen with this argument.

$$\text{gmap} : \{k : \text{Kind}\} \rightarrow (t : \text{Ty } k)$$

$$\rightarrow \text{Map } \langle k \rangle ([t] :: [t] :: [])$$

$$\text{gmap } t = \text{ngen } t \text{ gmap-const}$$

Providing the type code instantiates generic map at particular types. For example, using the code for the Option type of the previous section, we can define:

$$\begin{aligned} \text{option-map1} &: \{A \ B : \text{Set}\} \rightarrow (A \rightarrow B) \\ &\rightarrow \text{Option } A \rightarrow \text{Option } B \\ \text{option-map1} &= \text{gmap option} \end{aligned}$$

3.3 Doubly-generic Map

To use ngen to implement a doubly-generic function, we must also supply b and ce to ngen, but this time both of those arguments must generalize over the arity. For doubly-generic map, we call these pieces NGmap and ngen-const. NGmap is simply the arrTy function from Section 2.2, which takes the arity as an implicit argument.

$$\begin{aligned} \text{NGmap} &: \{n : \mathbb{N}\} \rightarrow \text{Vec Set } (\text{suc } n) \rightarrow \text{Set} \\ \text{NGmap} &= \text{arrTy} \end{aligned}$$

We are simplifying the example somewhat because generic zips (which are generic maps at arities greater than one) are partial functions—they may fail if instantiated at a sum type and passed mismatched injections. To account for this possibility in Generic Haskell, the library function zipWith returns a Maybe. However, we would like to keep our presentation as simple as possible, so we use an error term to indicate failure. A version of doubly-generic map that returns a Maybe is included with our sources. Because

Agda lacks Haskell's error function, we use a postulate:

postulate error : (A : Set) → A

Next, we define the behavior of arity-generic type-generic map at the constant types. We do this by writing a term that dispatches to cases for the various constants (defined below). Each case takes the arity as an argument.

```
ngmap-const : {n : ℕ} → TyConstEnv NGmap
ngmap-const {n} Nat   = defNat n
ngmap-const {n} Unit  = defUnit n
ngmap-const {n} Prod  = defPair n
ngmap-const {n} Sum   = defSum n
```

Recalling the definition of NGmap, for the first two cases, we must return arity- n functions with the types $\mathbb{N} \rightarrow \mathbb{N} \rightarrow \dots \rightarrow \mathbb{N}$ and $\top \rightarrow \top \rightarrow \dots \rightarrow \top$. For \mathbb{N} , when n is zero, we pick a default element to return arbitrarily. When n is one, we return the argument. For larger arities, we check that the inputs are identical. We choose to reject unequal nats to mirror the behavior of map at sum types.

```
defNat : (n : ℕ) → NGmap ⟨ ★ ⟩ (repeat {suc n} ℕ)
defNat zero      = zero      -- arbitrary ℕ
defNat (suc zero) = λ x → x  -- return what was given
defNat (suc (suc n)) =
  λ x → λ y → if eqNat x y then defNat (suc n) y
                else error _

defUnit : (n : ℕ) → NGmap ⟨ ★ ⟩ (repeat {suc n} ⊤)
defUnit zero     = tt
```

$$\text{defUnit } (\text{suc } n) = \lambda x \rightarrow (\text{defUnit } n)$$

The Prod and Sum cases remain. Because these constants have higher kinds, the return type of `ngmap-const` changes. Consider Prod first. The desired type of `defPair n` is:

$$\begin{aligned} \text{NGmap } \langle \star \Rightarrow \star \Rightarrow \star \rangle (\text{repeat } \{\text{suc } n\} _ \times _) = \\ \forall \Rightarrow \lambda (\text{As} : \text{Vec Set } n) \rightarrow \text{arrTy As} \rightarrow \\ \forall \Rightarrow \lambda (\text{Bs} : \text{Vec Set } n) \rightarrow \text{arrTy Bs} \rightarrow \\ \text{arrTy } ((\text{repeat } \times) \circledast \text{As} \circledast \text{Bs}) \end{aligned}$$

If we imagine writing out `As` as `A1 :: A2 :: ... :: An :: []` and `Bs` as `B1 :: B2 :: ... :: Bn :: []` the type simplifies to:

$$\begin{aligned} \text{NGmap } \langle \star \Rightarrow \star \Rightarrow \star \rangle (\text{repeat } \{\text{suc } n\} _ \times _) = \\ \{ \text{A1 A2 ... An} : \text{Set} \} \rightarrow (\text{A1} \rightarrow \text{A2} \rightarrow \dots \rightarrow \text{An}) \\ \rightarrow \{ \text{B1 B2 ... Bn} : \text{Set} \} \rightarrow (\text{B1} \rightarrow \text{B2} \rightarrow \dots \rightarrow \text{Bn}) \\ \rightarrow (\text{A1} \times \text{B1}) \rightarrow (\text{A2} \times \text{B2}) \rightarrow \dots \rightarrow (\text{An} \times \text{Bn}) \end{aligned}$$

However, it is easier to define the case where the `A1 ... An` arguments are uncurried, and then curry the resulting function.

```
defPairAux : (n : ℕ)
  → (As : Vec Set (suc n)) → arrTy As
  → (Bs : Vec Set (suc n)) → arrTy Bs
  → arrTy (repeat _ × _ ∘ledast As ∘ledast Bs)
defPairAux zero (A :: []) a (B :: []) b = (a, b)
defPairAux (suc n) (A1 :: As) a (B1 :: Bs) b =
  λ p →
    (defPairAux n As (a (proj₁ p)) Bs (b (proj₂ p)))
```

In the zero case of `defPairAux`, `a` and `b` are arguments of type `A`

and B respectively—the function must merely pair them up. In the successor case, a and b are functions with types $A1 \rightarrow \text{arrTy } As$ and $B1 \rightarrow \text{arrTy } Bs$. We want to produce a result of type $A1 \times B1 \rightarrow \text{arrTy}$ (repeat $_ \times _ \text{ } \circledast As \text{ } \circledast Bs$). Therefore, this case takes an argument p and makes a recursive call, passing in a applied to the first component of p and b applied to the second component of p . We use a kind-directed currying function $k\text{-curry}$, whose definition has been elided, to define the final version.

```
defPair : (n : ℕ)
  → NGmap ⟨  $\star \Rightarrow \star \Rightarrow \star$  ⟩ (repeat {suc n}  $\_ \times \_$ )
defPair n = k-curry ( $\star \Rightarrow \star \Rightarrow \star$ ) (defPairAux n)
```

Sum also has kind $\star \Rightarrow \star \Rightarrow \star$, so the type of its `ngmap-const` case is similar. However, for sums, we must check that the terms provided have the same structure (are either all inj_1 or all inj_2). Otherwise, we signal an error. Again, we first define `defSumAux` and then curry the result.

Below, `defSumAux` checks if the first argument is an inj_1 or an inj_2 , then calls `defSumLeft` or `defSumRight` which require that all subsequent arguments match. In the degenerate case, where there are no arguments, we arbitrarily choose the right injection. Because `defSumRight` is analogous to `defSumLeft`, we elide its definition below.

```
defSumLeft : (n : ℕ)
  → (As : Vec Set (suc n)) → arrTy As
  → (Bs : Vec Set (suc n))
  → arrTy (repeat  $\_ \sqcup \_ \text{ } \circledast As \text{ } \circledast Bs$ )
defSumLeft zero (A :: []) a (B :: []) = inj1 a
```

```

defSumLeft (suc n) (A1 :: As) a (B1 :: Bs) = f
  where
    f : A1  $\uplus$  B1  $\rightarrow$  arrTy (repeat  $\_ \uplus \_$   $\circledast$  As  $\circledast$  Bs)
    f (inj1 a1) = defSumLeft n As (a a1) Bs
    f (inj2 b1) = defSumLeft n As (a (error A1)) Bs

```

```

defSumAux : (n :  $\mathbb{N}$ )
 $\rightarrow$  (As : Vec Set (suc n))  $\rightarrow$  arrTy As
 $\rightarrow$  (Bs : Vec Set (suc n))  $\rightarrow$  arrTy Bs
 $\rightarrow$  arrTy (repeat  $\_ \uplus \_$   $\circledast$  As  $\circledast$  Bs)
defSumAux zero (A :: []) a (B :: []) b =
  (inj2 b)
defSumAux (suc n) (A1 :: As) a (B1 :: Bs) b = f
  where
    f : A1  $\uplus$  B1  $\rightarrow$  arrTy (repeat  $\_ \uplus \_$   $\circledast$  As  $\circledast$  Bs)
    f (inj1 a1) = defSumLeft n As (a a1) Bs
    f (inj2 b1) = defSumRight n As Bs (b b1)

```

Finally, we also curry `defSumAux` to get the desired branch.

```

defSum : (n :  $\mathbb{N}$ )
 $\rightarrow$  NGmap  $\langle \star \Rightarrow \star \Rightarrow \star \rangle$  (repeat {suc n}  $\_ \uplus \_$ )
defSum n = k-curry  $\langle \star \Rightarrow \star \Rightarrow \star \rangle$  (defSumAux n)

```

We can then define `ngmap` by instantiating `ngen`.

```

ngmap : (n :  $\mathbb{N}$ )  $\rightarrow$  {k : Kind}  $\rightarrow$  (e : Ty k)
 $\rightarrow$  NGmap  $\langle k \rangle$  (repeat {suc n}  $\_ \sqcup \_$ )
ngmap n e = ngen e ngmap-const

```

If we had included vectors in our universe, we could simply use

nmap from Section 2.2 for that case.

Just as datatype-generic functions are instantiated at a type, doubly generic functions are instantiated at an arity and a type. For example, given the definitions `Option` and `option` from the last section, we can define various maps for this type constructor:

```
option-map1 : { A B : Set } → (A → B)
              → Option A → Option B
option-map1 = ngmap 1 option
option-map2 : { A B C : Set } → (A → B → C)
              → Option A → Option B → Option C
option-map2 = ngmap 2 option
```

Of course, pair map functions are also instances of `ngmap`. We only show the definition of `pair-map2` below.

```
pair-map2 : { A1 B1 C1 A2 B2 C2 : Set }
            → (A1 → B1 → C1) → (A2 → B2 → C2)
            → A1 × A2 → B1 × B2 → C1 × C2
pair-map2 f1 f2 = ngmap 2 (Con Prod) f1 f2
```

4. Datatype Isomorphisms

The infrastructure described so far permits us to instantiate arity-generic functions at different types based on their structure. However, to complete the story and generate versions of `n-ary map` for datatypes like `Vec`, we must make a connection between arbitrary datatypes and their structure. In this section, we describe modifications to the implementation necessary to support generic functions on arbitrary datatypes through datatype isomorphisms.

4.1 Representing Datatypes

There are at least two ways to support datatypes. The current system already can encode datatypes on an ad hoc basis, in a manner described by Verbruggen et al. [26]. However, this encoding requires some tedious applications of coercions between the datatype and its isomorphism for each datatype instance of the generic operation. Instead, we move that boilerplate to the generic function itself by adding a new constructor to the `Typ` universe. This new constructor, `Data`, contains information about a particular datatype.

```
data Typ : Ctx → Kind → Set where
```

```
...
```

```
Data : ∀ {G} → DT G → Typ G ★
```

The `DT` data structure contains four pieces of information about a datatype: its `Typ` representation `t`, the actual Agda datatype that this code represents `s`, and two functions for coercing between values of type `t` and values of type `s`.

```
data DT (G : Ctx) : Set where
```

```
mkDT : (t      : Typ G ★)
```

```
      → (s      : Env G → Set)
```

```
      → (to      : ({e : Env G} → interp t e → s e))
```

```
      → (from    : ({e : Env G} → s e → interp t e))
```

```
      → DT G
```

Note that we can only represent datatypes of kind `Set`. Other kinds do not support the coercion functions `to` and `from` as their interpretations have the wrong type. To create isomorphisms of type constructors like `Vec`, the `DT` datatype is parameterized by a context `G`, and `s` may depend on an environment for that context.

We describe this mechanism in more detail below.

We define a number of accessor functions for retrieving the parts of a DT, called DT-s, DT-t, DT-from and DT-to (here elided).⁴ Because `interp` is mentioned by the components of DT, it must be defined mutually with `Env`, `sLookup`, `Typ`, `DT`, and its accessors.

Finally, we extend the interpretation function for codes by looking up the Agda type and giving it the current environment.

$$\text{interp} : \forall \{k\} G \rightarrow \text{Typ } G \, k \rightarrow \text{Env } G \rightarrow \llbracket k \rrbracket$$

...

$$\text{interp} (\text{Data } dt) \, e = \text{DT-s } dt \, e$$

For example, suppose we have a simple datatype definition that identifies natural numbers as Oranges.

⁴Unfortunately, Agda does not support record definitions in a mutual block.

data Orange : Set **where**

toOrange : $\mathbb{N} \rightarrow \text{Orange}$

We can form the code for this datatype as below.

$$\text{fromOrange} : \text{Orange} \rightarrow \mathbb{N}$$
$$\text{fromOrange} (\text{toOrange } x) = x$$
$$\text{orange} : \{G : \text{Ctx}\} \rightarrow \text{Typ } G \, \star$$
$$\text{orange} = \text{Data } (\text{mkDT}$$
$$(\text{Con } \text{Nat}) \quad \text{-- } t$$
$$(\lambda e \rightarrow \text{Orange}) \quad \text{-- } s$$
$$(\lambda \{e\} \rightarrow \text{toOrange}) \quad \text{-- } \text{to}$$
$$(\lambda \{e\} \rightarrow \text{fromOrange})) \quad \text{-- } \text{from}$$

Even though the kind of a datatype isomorphism must be \star , we

can still create isomorphisms for datatypes with higher kinds, such as `Maybe` and `Vec`. This works by creating an isomorphism with a "hole" (exploiting the fact that the environment need not be empty), then wrapping it in a `lambda`.

Instead of defining the structure of the `Maybe` type as a code with higher kind (i.e., something of type `Ty (★ ⇒ ★)`, such as `option` from Section 3.1), we instead define its structure as a function from codes to codes.

```
maybeDef : { G : Ctx } → Typ G ★ → Typ G ★
maybeDef t = (App (App (Con Sum) (Con Unit)) t)
```

The conversions to and from the `Maybe` type are also parameterized by the code of the argument to `Maybe`.

```
toMaybe : { G : Ctx } { e : Env G }
          → (t : Typ G ★)
          → (interp (maybeDef t) e) → Maybe (interp t e)
toMaybe t (inj1 _) = nothing
toMaybe t (inj2 x) = just x
fromMaybe : { G : Ctx } { e : Env G } → (t : Typ G ★)
          → Maybe (interp t e)
          → (interp (maybeDef t) e)
fromMaybe t (just x) = inj2 x
fromMaybe t nothing = inj1 tt
```

Finally, we form the code of the datatype itself by wrapping the `Data` constructor in a `Lam` and using variable `zero` for the parameter. The environments supplied to `s`, `to` and `from` allow us to specify the type this variable corresponds to.


```

maybe : { G : Ctx } → Typ G (★ ⇒ ★)
maybe { G } = Lam (Data
  (mkDT (maybeDef (Var VZ))
    (λ e → Maybe (interp (Var VZ) e))
    (λ { e } → toMaybe {★ :: G} { e } (Var VZ))
    (λ { e } → fromMaybe {★ :: G} { e } (Var VZ))))))

```

We can use this same idea to encode vectors. Because we know the length of given vector, it is isomorphic to an n-tuple—a sequence of products terminated by unit. The code for the vector type then abstracts both the code for the type of the elements of the vector and a natural number for its length.

```

vecDef : { G : Ctx } → Typ G ★ → (n : ℕ) → Typ G ★
vecDef _ 0 = Con Unit
vecDef t (suc n) =
  (App (App (Con Prod) t) (vecDef t n))
fromVec : { n : ℕ } { G : Ctx }
  { t : Typ G ★ } { e : Env G }
  → Vec (interp t e) n → (interp (vecDef t n) e)
fromVec { 0 } [] = tt
fromVec { suc n } (x :: xs) = (x, fromVec xs)
toVec : { n : ℕ } { G : Ctx } { t : Typ G ★ } { e : Env G }
  → interp (vecDef t n) e → Vec (interp t e) n
toVec { 0 } _ = []
toVec { suc n } (x, xs) = (x :: toVec xs)
vec : { G : Ctx } → { n : ℕ } → Typ G (★ ⇒ ★)
vec { G } { n } = Lam (Data
  (mkDT (vecDef (Var VZ) n)

```

$$(\lambda e \rightarrow \text{Vec} (\text{interp} (\text{Var } VZ) e) n) \\ \text{toVec} \\ \text{fromVec}))$$

Lists are somewhat trickier to represent. The type of a list does not tell us its length, so the direct recursive representation of `List` is an infinite structure. Agda’s termination checker will be unable to prove that uses of this representation terminate. Another option is to encode dependent pairs of lists and proofs they have finite lengths, rather than encoding lists directly. Examples of both these encodings are included with our source code.

4.2 Adding Data Support to `ngen`

Although `Data` provides a mechanism for coding datatypes, we cannot use it to define generic functions until we extend `ngen` to handle `Data`. However, there is a complication—it is not clear how to do so. Without getting too much into the technicalities, the issue is that in this definition we need to produce a result of type⁵

$$b \text{ (repeat (DT-s dt) } \circledast \text{ envs)}$$

but we only have a value of type

$$b \text{ (repeat (interp (DT-t dt)) } \circledast \text{ envs)}$$

We would like to coerce the latter to the former using `to` and `from`, but we know nothing about `b`. Therefore, we require an additional argument to `ngen`, to be supplied when the generic operation is defined (i.e., when `b` is supplied).

$$\text{ngen} : \{n : \mathbb{N}\} \{b : \text{Vec Set} (\text{suc } n) \rightarrow \text{Set}\} \{k : \text{Kind}\} \\ \rightarrow (t : \text{Ty } k)$$

$\rightarrow \text{TyConstEnv } n \ b$
 $\rightarrow \text{DataGen } b$
 $\rightarrow b \langle k \rangle (\text{repeat } \lfloor t \rfloor)$

This argument, of type `DataGen`, shown below, is exactly the coercion function necessary.

```

DataGen : { n : ℕ } → ( b : Vec Set (suc n) → Set ) → Set
DataGen { n } b =
  { G : Ctx }
  → ( dt : DT G )
  → ( envs : Vec (Env G) (suc n) )
  → b (repeat (interp (DT-t dt)) ⊗ envs)
  → b (repeat (DT-s dt) ⊗ envs)

```

As an example of an instance of `DataGen`, recall the definition of `ngmap` and its base type `NGmap` from Section 3.3.

```

arrTy : { n : ℕ } → Vec Set (suc n) → Set
arrTy { 0 } (A :: []) = A
arrTy { suc n } (A1 :: As) = A1 → arrTy As

NGmap : { n : ℕ } → Vec Set (suc n) → Set
NGmap = arrTy

```

The definition of the `DataGen` coercion for the case where `b` is `NGmap`, called `ngmap-data` below, proceeds by induction on

⁵ Here `b : Vec Set (suc n) → Set` describes the type of the generic operation and `envs : Vec (Env G) (suc n)` is a vector of environments for the free variables.

the arity. In the base case of `n=0`, `ngmap-data` must coerce a result from the representation type to the Agda type using the `to`

component.

For higher n , `ngmap-data` is provided with a vector of environments $e1 :: es$ and a function of type:

$$\begin{aligned} & \text{interp (DT-t dt) } e1 \\ & \rightarrow \text{arrTy (repeat (interp (DT-t dt)) } \circledast \text{ es)} \end{aligned}$$

Its result type is:

$$(\text{DT-s dt}) e1 \rightarrow \text{arrTy (repeat (DT-s dt) } \circledast \text{ es)}$$

This case takes in a $(\text{DT-s dt}) e1$, uses the `from` function to convert it to an $\text{interp (DT-t dt) } e1$, then coerces the result of the provided function by calling `ngmap-data` recursively.

$$\begin{aligned} \text{ngmap-data} & : \{n : \mathbb{N}\} \rightarrow \text{DataGen (NGmap } \{n\}) \\ \text{ngmap-data } \{0\} \quad dt (e :: []) \text{ bt} & = \text{DT-to dt bt} \\ \text{ngmap-data } \{\text{suc } n\} \text{ dt } (e1 :: es) \text{ bt} & = \\ & \lambda x \rightarrow \text{ngmap-data } \{n\} \text{ dt } es (\text{bt (DT-from dt } x)) \end{aligned}$$

4.3 Using `ngen` at Datatypes

With the `ngmap-data` function from the previous section, we may instantiate the updated `ngen` for `NGmap`.

$$\begin{aligned} \text{ngmap} & : (n : \mathbb{N}) \rightarrow \{k : \text{Kind}\} \rightarrow (e : \text{Typ } [] k) \rightarrow \\ & \quad \text{NGmap } \{n\} \langle k \rangle (\text{repeat (interp } e [])) \\ \text{ngmap } n \text{ e} & = \text{ngen } e \text{ ngmap-const ngmap-data} \end{aligned}$$

This new `ngmap` adds support for datatypes. For example, we may use it with the `maybe` and `vec` representations of Section 4.1. Note that `vec-map0` is precisely the `repeat` function we have used throughout this paper.

```

maybe-map1 : {A B : Set} → (A → B)
              → Maybe A → Maybe B
maybe-map1 = ngmap 1 maybe

vec-map0    : {A : Set} {n : ℕ} → A → Vec A n
vec-map0    = ngmap 0 vec

vec-map1    : {A B : Set} {n : ℕ}
              → (A → B) → Vec A n → Vec B n
vec-map1    = ngmap 1 vec

```

Observe that instantiating `ngmap` at a datatype is no different than any other type we have seen. The codes for `Maybe` and `Vec` work for any generic operation. Although the definition of `ngmap` needed the `DataGen` argument, this argument must be implemented once per generic operation, just like `TyConstEnv`. In contrast, previous work [26] could not define a general code for datatypes like `Maybe` and `Vec`, and required significant boilerplate at every instantiation of a generic function with a specific datatype.

5. Other Doubly-Generic Operations

Mapping is not the only arity-generic function. In this section, we examine two others.

5.1 Equality

We saw in Section 3.3 that doubly-generic `map` must check that its arguments have the same structure. We can define doubly-generic equality in a similar manner. This function takes `n` arguments, returning `true` if they are all equal, and `false` otherwise. Unlike `map`, equality is not partial for sums as it returns `false` in the case that the

injections do not match.

In the specific case of vectors, arity-generic equality looks a lot like arity-generic map. Each instance of this function follows the same pattern. Given an n-ary equality function for the type argument, we can define n-ary equality for vectors as:

$$\begin{aligned}\text{nvec-eq} &: \{m : \mathbb{N}\} \{A : \text{Set}\} \\ &\rightarrow (A \rightarrow \dots \rightarrow A \rightarrow \text{Bool}) \\ &\rightarrow \text{Vec } A \ m \rightarrow \dots \rightarrow \text{Vec } A \ m \rightarrow \text{Bool} \\ \text{nvec-eq } f \ v1 \ \dots \ vn &= \text{all } (\text{repeat } f \circledast v1 \circledast \dots \circledast vn)\end{aligned}$$

However, again this definition does not help us make equality type-generic as well as arity-generic. For type-genericity, the type of the equality function depends on the kind of the type constructor.

For example, the definition of arity-two equality for natural numbers returns true only if all three match:

$$\text{nat-eq2} : \mathbb{N} \rightarrow \mathbb{N} \rightarrow \mathbb{N} \rightarrow \text{Bool}$$

Likewise, the arity-two equality for pairs requires equalities for all of the components of the pair. Furthermore, the type arguments need not be the same. We can pass any sort of comparison functions in to examine the values carried by the three products.

$$\begin{aligned}\text{pair-eq2} &: \{A1 \ B1 \ C1 \ A2 \ B2 \ C2 : \text{Set}\} \\ &\rightarrow (A1 \rightarrow B1 \rightarrow C1 \rightarrow \text{Bool}) \\ &\rightarrow (A2 \rightarrow B2 \rightarrow C2 \rightarrow \text{Bool}) \rightarrow \\ &\rightarrow (A1 \times A2) \rightarrow (B1 \times B2) \rightarrow (C1 \times C2) \rightarrow \text{Bool}\end{aligned}$$

The definition of `ngeq`, which can define all of these operations, is similar to that of `ngmap`, so we will only highlight the differences.⁶ One occurs in the definition of the arity-indexed type, `NGeq`. This function returns a boolean value rather than one of

the provided types, which means that `ngeq` makes sense even for `n = 0`. In that case its type is simply `Bool`.

```
NGeq : {n : ℕ} → (v : Vec Set n) → Set
NGeq {zero} [] = Bool
NGeq {suc n} (A1 :: As) = A1 → NGeq As
```

Next we must define a `TyConstEnv` for `NGeq`. For simplicity, we only show the cases for `Unit` and `Nat`. The cases for `Prod` and `Sum` are straightforward variations of `ngmap`. As there is only a single member of the \top type, the case for `unit` is just a function that takes `n` arguments and returns `true`.

```
defUnit : (n : ℕ) → NGeq (repeat ⊤)
defUnit zero = λ x → true
defUnit (suc n) = λ x → defUnit n
```

For natural numbers, `ngeq` should compare each number and return `true` only when they all match (or when `n` is less than 2). We implement this by checking each argument for equality with the next. If a mismatch is found, `ngeq` uses `constFalse`, which consumes a given number of arguments and returns `false`.

```
constFalse : {n : ℕ} → (v : Vec Set n) → NGeq v
constFalse {zero} [] = false
constFalse {suc m} (A1 :: As) = λ a → constFalse As
defNat : (n : ℕ) → NGeq (repeat {n} ℕ)
defNat zero = true
defNat (suc zero) = λ x → true
defNat (suc (suc n)) =
  λ x → λ y → if eqNat x y then defNat (suc n) y
               else constFalse (repeat ℕ)
```

Finally, because we wish to use `ngeq` at various Agda datatypes, we must define an instance of `DataGen` from Section 4. As before, we go by recursion on the arity. Since `NGeq` is an n -ary function of representable types, we simply take in each argument, use the provided `DT` isomorphism to coerce it to the appropriate type, and recurse:

$$\begin{aligned} \text{ngeq-data} &: \{n : \mathbb{N}\} \rightarrow \text{DataGen } (\text{NGeq } \{\text{suc } n\}) \\ \text{ngeq-data } \{0\} \quad dt \ (e :: []) \ bt &= \end{aligned}$$

⁶The complete definition may be found in `ngeq.agda` with our sources.

$$\begin{aligned} &\lambda s \rightarrow bt \ (DT\text{-from } dt \ s) \\ \text{ngeq-data } \{\text{suc } n\} \ dt \ (e :: es) \ bt &= \\ &\lambda s \rightarrow \text{ngeq-data } dt \ es \ (bt \ (DT\text{-from } dt \ s)) \end{aligned}$$

With these pieces defined, the definition of `ngeq` is a straightforward application of `ngen`:

$$\begin{aligned} \text{ngeq} &: (n : \mathbb{N}) \rightarrow \{k : \text{Kind}\} \rightarrow (e : \text{Ty } k) \rightarrow \\ &\quad \text{NGeq } \langle k \rangle \ (\text{repeat } \{\text{suc } n\} \ (\lambda e \])) \\ \text{ngeq } n \ e &= \text{ngen } e \ \text{ngeq-const } \text{ngeq-data} \end{aligned}$$

5.2 Splitting

The Haskell prelude and standard library include the functions

$$\begin{aligned} \text{unzip} &:: [(a, b)] && \rightarrow ([a], [b]) \\ \text{unzip3} &:: [(a, b, c)] && \rightarrow ([a], [b], [c]) \\ \text{unzip4} &:: [(a, b, c, d)] && \rightarrow ([a], [b], [c], [d]) \\ \text{unzip5} &:: [(a, b, c, d, e)] && \rightarrow ([a], [b], [c], [d], [e]) \\ \text{unzip6} &:: [(a, b, c, d, e, f)] && \rightarrow ([a], [b], [c], [d], [e], [f]) \end{aligned}$$

suggesting that there should be an arity-generic version of `unzip` that unifies all of these definitions.

Furthermore, it makes sense that we should be able to unzip data structures other than lists, such as `Maybes` or `Trees`.

$$\begin{aligned}\text{unzipMaybe} &:: \text{Maybe } (a, b) \rightarrow (\text{Maybe } a, \text{Maybe } b) \\ \text{unzipTree} &:: \text{Tree } (a, b) \rightarrow (\text{Tree } a, \text{Tree } b)\end{aligned}$$

Indeed, `unzip` is also datatype-generic, and `Generic Haskell` includes the function `gunzipWith` that can generate arity-one unzips for any type constructor.

Here, we describe the definition of `ngsplit`, which generates unzips for arbitrary data structures at arbitrary arities. In some sense, `ngsplit` is the dual to `ngmap`. Instead of taking in n arguments (with the same structure) and combining them together to a single result, `split` takes a single argument and distributes it to n results, all with the same structure.

For example, here is an instance of `ngsplit`, specialized to the `Option` type and arity 2. Note that this function is more general than `unzipMaybe` above, the `Maybes` need not contain pairs so long as we have some way to split the data.

$$\begin{aligned}\text{unzipWithMaybe2} &: \{A\ B\ C : \text{Set}\} \rightarrow (A \rightarrow B \times C) \\ &\rightarrow (\text{Maybe } A \rightarrow \text{Maybe } B \times \text{Maybe } C) \\ \text{unzipWithMaybe2} &= \text{ngsplit } 2\ \text{maybe}\end{aligned}$$

The definition of `unzipWith` gives us `unzip` when applied to the identity function.

$$\begin{aligned}\text{unzipMaybe2} &: \{A\ B : \text{Set}\} \rightarrow \text{Maybe } (A \times B) \\ &\rightarrow (\text{Maybe } A \times \text{Maybe } B)\end{aligned}$$

$\text{unzipMaybe2} = \text{unzipWith2 } (\lambda x \rightarrow x)$

The function `NGsplit` gives the type of `ngsplit` at base kinds. The first type in the vector passed to `NGsplit` is the type to split. The subsequent types are those the first type will be split into. If there is only one type, the function returns `unit`. The helper function `prodTy` folds the `_ × _` constructor across a vector of types.

```
prodTy : { n : ℕ } → (As : Vec Set n) → Set
prodTy {0} _ = ⊤
prodTy {1} (A :: []) = A
prodTy {suc (suc _)} (A :: As) = (A × prodTy As)
```

```
NGsplit : { n : ℕ } → (v : Vec Set (suc n)) → Set
NGsplit (A1 :: As) = A1 → prodTy As
```

The cases for `Nat` and `Unit` are straightforward, so we do not show them. They simply make `n` copies of the argument.

To split a product (x, y) , we first split x and y , then combine together the results. For this combination, `prodn` takes arguments of types $(A1 \times A2 \times \dots \times An)$ and $(B1 \times B2 \times \dots \times Bn)$ and forms a result of type $(A1 \times B1) \times (A2 \times B2) \times \dots \times (An \times Bn)$.

```
prodn : { n : ℕ } → (As Bs : Vec Set n)
      → prodTy As → prodTy Bs
      → prodTy (repeat _ × _ ⊗ As ⊗ Bs)
prodn {0} _ _ a b = tt
prodn {1} (A :: []) (B :: []) a b = (a, b)
prodn {suc (suc n)} (A :: As) (B :: Bs) (a, as) (b, bs) =
  ((a, b), prodn {suc n} _ _ as bs)
```

```

defPair : (n : ℕ)
  → (As : Vec Set (suc n)) → (NGsplit As)
  → (Bs : Vec Set (suc n)) → (NGsplit Bs)
  → NGsplit (repeat _ × _ ⊗ As ⊗ Bs)
defPair n (A :: As) a (B :: Bs) b =
  λ p → prodn {n} _ _ (a (proj1 p)) (b (proj2 p))

```

The case for sums scrutinizes the argument to see if it is a left or right injection, and uses the appropriate provided function to split the inner expression. Then we use either `injLeft` or `injRight` (elided), which simply map `inj1` or `inj2` onto the members of the resulting tuple.

```

defSum : (n : ℕ)
  → (As : Vec Set (suc n)) → (NGsplit As)
  → (Bs : Vec Set (suc n)) → (NGsplit Bs)
  → NGsplit (repeat _ ⊕ _ ⊗ As ⊗ Bs)
defSum 0 (A :: []) af (B :: []) bf = λ _ → tt
defSum (suc n) (A :: As) af (B :: Bs) bf = f
  where f : A ⊕ B → prodTy (repeat _ ⊕ _ ⊗ As ⊗ Bs)
        f (inj1 x1) = injLeft {n} (af x1)
        f (inj2 x1) = injRight {n} (bf x1)

```

The definition of `split-const` (elided) dispatches to the branches above in the standard way, delegating to a trivial case when `n` is 0. Finally, we must define an instance of `DataGen` so that we may use `ngsplit` at representable Agda datatypes. Since `NGsplit` is defined in terms of `prodTy`, we must also convert instances of that type. These (elided) functions are similar to previous examples, except that we are converting a pair instead of an arrow. With `split-const`

and `split-data`, we can define `ngsplit` as usual.

Splitting is a good example of datatype-generic programming's potential to save time and eliminate errors. Defining a separate instance of `split` for vectors is not simple. For example, we would need a function to transpose vectors of products, transforming $\text{Vec } m \ (A_1 \times A_2 \times \dots \times A_n)$ into $(\text{Vec } A_1 \ m \times \text{Vec } A_2 \ m \times \dots \times \text{Vec } A_n \ m)$. This code is slightly tricky and potentially error-prone, but with generic programming we get the vector `split` for free. Moreover, we may reason once about the correctness of the general definition of `split` rather than reasoning individually about each of its arity and type instances.

5.3 More Operations

Mapping, equality and splitting provide three worked out examples of doubly generic functions. We know of a few others, such as a monadic map, a map that returns a `Maybe` instead of an error when the `Sum` injections do not match, a comparison function, and an equality function that returns a proof that the arguments are all equal. Furthermore, there are arity-generic versions of standard Generic Haskell functions like `crushes` or `enumerations`. For example, an arity-generic `gsum` adds together all of the numbers found in `n` data structures. Such examples seem less generally useful than arity-generic `map` or `unzip`, but are not difficult to define.

Compared to the space of datatype-generic functions, the space of doubly generic operations is limited. This is unsurprising, as there already were not many examples of Generic Haskell functions with arities greater than one. However, this work has given us new insight into what other doubly-generic functions might look like. Furthermore, though the collection of doubly-generic functions is small, this is no reason not to study it. Indeed, it includes some of

the most fundamental operations of functional programming, and it makes sense that we should learn as much as we can about these operations.

6. Related Work

Only a few sources discuss arity-generic programming. Fridlender and Indrika [7] show how to encode n -ary list map in Haskell, using a Church encoding of numerals to reflect the necessary type dependencies. They remark that a generic programming language could provide a version of `zipWith` that works for arbitrary datatypes, but that no existing language provides such functionality. They also mention a few other arity-generic programs: `taut` which determines whether a boolean expression of n variables is a tautology, and variations on `liftM`, `curry` and `uncurry` from the Haskell prelude. It is not clear whether any of these functions could be made datatype-generic. McBride [14] shows an alternate encoding of arity-generic list map in Haskell using type classes to achieve better safety properties. He examines several other families of operations, like `crush` and `sum`, but does not address type genericity.

Many Scheme functions, such as `map`, are arity-generic (or variable-arity, in Scheme parlance). Strickland et al. [24] extend Typed Scheme with support for variable-arity polymorphism by adding new forms for variable-arity functions to the type language. They are able to check many examples, but do not consider datatype-genericity.

Sheard [22] translates Fridlender and Indrika's example to the Ω mega programming language, using that language's native indexed datatypes instead of the Church encoding. He also demonstrates one other arity-generic program, n -ary addition. Although

the same work also includes an implementation of datatype-generic programming in Ω mega, the two ideas are not combined.

Several researchers have used dependent types (or their encodings) to implement Generic-Haskell-style datatype-genericity. In previous work, we encoded representations of types using Church encodings [28] and GADTs [29] and showed how to implement a number of datatype-generic operations such as `map`. Hinze [10], inspired by this approach, gave a similar encoding based on type classes. In those encodings, doubly-generic programming is not possible because datatype-generic programs of different arities require different representations or type classes.

The most closely related encoding of Generic Haskell to this one is by Verbruggen et al. [26]. They use the Coq programming language to define a framework for generic programming, but do not consider arity-genericity. Altenkirch and McBride [1] show a similar development in Oleg. This work extends those developments by considering examples not possible in Generic Haskell and showing a technique for writing generic programs which work on source-language datatypes.

The idea of generic programming in dependent type theory via universes has seen much attention since it was originally proposed [13, 18]. While demonstrating a new form of double genericity, this paper covers only one part of what is possible in a dependently typed language. In particular, our codes do not extend to all inductive families and so we cannot represent all types that are available (see Benke et al. [2] and Morris et al. [17] for more expressive universes). A dependently-typed language also permits the definition of generic proofs about generic programs. Chlipala [4] uses this technique in the Coq proof assistant to generically define and prove substitution properties of programming languages. Ver-

bruggen et al. [27] use Coq’s dependent types to develop a framework for proving properties about generic programs.

7. Discussion

Generic programming in a dependently-typed language As we mentioned in the introduction, there are several dependently-typed languages that we could have used for this development. We selected Agda because the focus of its design has been this sort of programming. Like Coq, Agda is a full-spectrum dependently typed language. That has allowed us the flexibility to use universes to directly implement generic programming. We had the full power of the computational language available to express the relationships between values and types. A phase-sensitive language, such as Ω mega or Haskell, would have required singletons to reflect computation to the type level, and would have permitted type-level computation only in a restricted language.

Compared to Coq, Agda has more vigorous type inference, especially combined with pattern matching. Although Coq can also infer implicit arguments, if we had written the functions in Coq we would have had to add many more type annotations. Additionally, developing in Agda allowed us to deal with non-termination more conveniently—while Coq must be able to see that a definition terminates before moving on, Agda shows the user where it can not prove termination and allows other work to continue.

On the other hand, using Coq would have lead to two advantages. Coq’s tactic language can be used to automate some of the reasoning. Tactics would have been particularly useful in proving some of the equalities needed to typecheck the (elided) implementation of `ngen`. However, we did not see any need for tactics in any

of the *uses* of `ngen` to define doubly-generic operations. More importantly, as discussed below, differences in the way `Coq` and `Agda` handle type levels forced us to use `Agda`'s `--type-in-type` flag to clarify the presentation.

Type levels in Agda Although we have hidden it, `Agda` actually has an infinite hierarchy of type levels. `Set`, also known as `Set0`, is the lowest level in the type hierarchy. Terms like `Set0` and `Set0 → Set0` have type `Set1`, which itself has type `Set2`, etc.

To simplify our exposition, we collapsed all of these levels to the type `Set`, with the help of the `--type-in-type` flag. This flag makes `Agda`'s logic inconsistent, so to demonstrate that we are not using it in an unsound way, we have also implemented a version of the code that may be compiled without the flag. That version can be found in the `notypeintype` subdirectory of our source tarball.

Three differences between `Coq` and `Agda` make this explicit version more complicated than the one presented here. First, `Agda` currently lacks *universe polymorphism* [9], a feature which allows definitions to work on multiple type levels. As a result, many of the data structures in this paper must actually be duplicated at the level of `Set1`, creating significant clutter. Second, since `Set` is not impredicative in `Agda`, many definitions that could live at the level of `Set` in `Coq` must be at the level of `Set1` instead. Finally, because `Set0` is not a subtype of `Set1` in `Agda`, we found it necessary to explicitly lift types from `Set0` to `Set1`.

Future work and Conclusions Because we are working in the flexible context of a dependently-typed programming language, our work here will allow us to adapt and extend orthogonal results in generic programming to this framework. For example, we would like to use `Agda` as a proof assistant to reason about the properties

of the generic programs that we write. We would also like to extend our universe so that it may encode more of Agda’s type system, such as arbitrary indexed datatypes. Finally, we would like to gain more experience with doubly-generic programming by creating and analyzing additional examples.

In this paper, we have combined arity-generic and datatype-generic programming into a single framework. Crucially, this combination takes advantage of the natural role that arities play in the definition of kind-indexed types. This framework has provided us with new understanding of the definition and scope of doubly-generic programs.

Acknowledgments Thanks to Andres Löb and Tim Sheard for discussion, and to the anonymous reviewers for many helpful comments. This paper was generated from literate Agda sources using `lhs2TeX`.

References

- [1] Thorsten Altenkirch and Conor McBride. Generic programming within dependently typed programming. In *Proceedings of the IFIP TC2 Working Conference on Generic Programming*, Dagstuhl, Germany, July 2003.
- [2] Marcin Benke, Peter Dybjer, and Patrik Jansson. Universes for generic programs and proofs in dependent type theory. *Nordic Journal of Computing*, 10(4):265–289, 2003.
- [3] Manuel M. T. Chakravarty, Gabriele Keller, and Simon Peyton Jones. Associated type synonyms. In *ICFP ’05: Proceedings of the Tenth ACM SIGPLAN International Conference on Functional Programming*, pages 241–253, New York, NY, USA, 2005. ACM.
- [4] Adam Chlipala. A certified type-preserving compiler from lambda

- calculus to assembly language. In *PLDI '07: Proceedings of the 2007 ACM SIGPLAN Conference on Programming Language Design and Implementation*, pages 54–65, New York, NY, USA, 2007. ACM.
- [5] Dave Clarke, Ralf Hinze, Johan Jeuring, Andres Löb, and Jan de Wit. The Generic Haskell user’s guide. Technical Report UU-CS-2001-26, Utrecht University, 2001.
 - [6] Bruno C. d. S. Oliveira, Ralf Hinze, and Andres Loeh. Extensible and modular generics for the masses. In Henrik Nilsson, editor, *Trends in Functional Programming (TFP 2006)*, April 2007.
 - [7] Daniel Fridlender and Mia Indrika. Do we need dependent types? *Journal of Functional Programming*, 10(4):409–415, July 2000.
 - [8] Jean-Yves Girard. *Interprétation fonctionnelle et élimination des coupures de l’arithmétique d’ordre supérieur*. PhD thesis, Université Paris VII, 1972.
 - [9] Robert Harper and Robert Pollack. Type checking with universes. *Theoretical Computer Science*, 89:107–136, 1991.
 - [10] Ralf Hinze. Generics for the masses. *Journal of Functional Programming*, 16(4-5):451–483, 2006.
 - [11] Ralf Hinze. Polytypic values possess polykinded types. *Science of Computer Programming*, 43(2–3):129–159, 2002. MPC Special Issue.
 - [12] Ralf Hinze and Johan Jeuring. Generic Haskell: Practice and theory. In Roland Backhouse and Jeremy Gibbons, editors, *Generic Programming, Advanced Lectures*, volume 2793 of *Lecture Notes in Computer Science*, pages 1–56. Springer-Verlag, 2003.
 - [13] Per Martin-Löf. *Intuitionistic Type Theory*. Bibliopolis-Napoli, 1984.
 - [14] Conor McBride. Faking it: Simulating dependent types in haskell. *Journal of Functional Programming*, 12(5):375–392, 2002.
 - [15] Conor McBride and James McKinna. The view from the left. *Journal of Functional Programming*, 14(1):69–111, 2004.

- [16] Conor McBride and Ross Paterson. Applicative programming with effects. *Journal of Functional Programming*, 18(1):1–13, 2008.
- [17] Peter Morris, Thorsten Altenkirch, and Neil Ghani. Constructing strictly positive families. In *CATS '07: Proceedings of the Thirteenth Australasian Symposium on Theory of Computing*, pages 111–121, Darlinghurst, Australia, Australia, 2007. Australian Computer Society, Inc.
- [18] B. Nordström, K. Petersson, and J. Smith. *Programming in Martin-Löf's Type Theory: an introduction*. Oxford University Press, 1990.
- [19] Ulf Norell. *Towards a practical programming language based on dependent type theory*. PhD thesis, Department of Computer Science and Engineering, Chalmers University of Technology, SE-412 96 Göteborg, Sweden, September 2007.
- [20] Simon Peyton Jones et al. The Haskell 98 language and libraries: The revised report. *Journal of Functional Programming*, 13(1):0–255, Jan 2003. <http://www.haskell.org/definition/>.
- [21] Simon L. Peyton Jones, Dimitrios Vytiniotis, Stephanie Weirich, and Geoffrey Washburn. Simple unification-based type inference for GADTs. In *ICFP '06: Proceedings of the Eleventh ACM SIGPLAN International Conference on Functional Programming*, pages 50–61, Portland, OR, USA, September 2006.
- [22] Tim Sheard. Generic programming programming in omega. In Roland Backhouse, Jeremy Gibbons, Ralf Hinze, and Johan Jeuring, editors, *Datatype-Generic Programming*, volume 4719 of *Lecture Notes in Computer Science*, pages 258–284. Springer, 2006.
- [23] Tim Sheard. Putting Curry-Howard to work. In *Proceedings of the ACM SIGPLAN 2005 Haskell Workshop*. ACM Press, September 2005.
- [24] T. Stephen Strickland, Sam Tobin-Hochstadt, and Matthias Felleisen. Practical variable-arity polymorphism. In *ESOP '09: Proceedings of*

the Eighteenth European Symposium On Programming, pages 32–46, March 2009.

- [25] The Coq Development Team. *The Coq Proof Assistant Reference Manual, Version 8.1*. LogiCal Project, 2006. Available from <http://coq.inria.fr/V8.1beta/refman/>.
- [26] Wendy Verbruggen, Edsko de Vries, and Arthur Hughes. Polytypic programming in Coq. In *WGP '08: Proceedings of the ACM SIGPLAN Workshop on Generic Programming*, pages 49–60, New York, NY, USA, 2008. ACM.
- [27] Wendy Verbruggen, Edsko de Vries, and Arthur Hughes. Polytypic properties and proofs in Coq. In *WGP '09: Proceedings of the 2009 ACM SIGPLAN Workshop on Generic Programming*, pages 1–12, New York, NY, USA, 2009. ACM.
- [28] Stephanie Weirich. Type-safe run-time polytypic programming. *Journal of Functional Programming*, 16(10):681–710, November 2006.
- [29] Stephanie Weirich. RepLib: A library for derivable type classes. In *Haskell Workshop*, pages 1–12, Portland, OR, USA, September 2006.