

# Effect Handlers in Scope

Nicolas Wu

University of Oxford

`nicolas.wu@cs.ox.ac.uk`

Tom Schrijvers

Ghent University

`tom.schrijvers@ugent.be`

Ralf Hinze

University of Oxford

`ralf.hinze@cs.ox.ac.uk`

## Abstract

Algebraic effect handlers are a powerful means for describing effectful computations. They provide a lightweight and orthogonal technique to define and compose the syntax and semantics of different effects. The semantics is captured by handlers, which are functions that transform syntax trees.

Unfortunately, the approach does not support syntax for scoping constructs, which arise in a number of scenarios. While handlers can be used to provide a limited form of scope, we demonstrate that this approach constrains the possible interactions of effects and rules out some desired semantics.

This paper presents two different ways to capture scoped constructs in syntax, and shows how to achieve different semantics by reordering handlers. The first approach expresses scopes using the existing algebraic handlers framework, but has some limitations. The problem is fully solved in the second approach where we introduce higher-order syntax.

***Categories and Subject Descriptors*** D.1.1 [*Programming Techniques*]: Functional Programming

***General Terms*** Languages

***Keywords*** Haskell, effect handlers, modularity, monads, syntax, semantics

# 1. Introduction

Effect handlers [13] have established themselves as a lightweight and compositional means of describing effectful computations. At the heart of the solution is the idea that a program is composed out of fragments of syntax that are often orthogonal to one another. Those fragments can then in turn be given a semantics by handlers that systematically deal with different effects.

One aspect of handlers that has not received much attention are scoping constructs. Examples of this abound: we see it in constructions for control flow, such as while loops and conditionals, but we also see it in pruning nondeterministic computations, exception handling, and multi-threading. The current work on effect handlers considers scoping to be in the province of handlers, that not only provide semantics but also delimit the scope of their effects.

However, as this paper illustrates, using handlers for scoping has an important limitation. The reason is that the semantics of handlers

Permission to make digital or hard copies of all or part of this work for personal or classroom use is granted without fee provided that copies are not made or distributed for profit or commercial advantage and that copies bear this notice and the full citation on the first page. Copyrights for components of this work owned by others than the author(s) must be honored. Abstracting with credit is permitted. To copy otherwise, or republish, to post on servers or to redistribute to lists, requires prior specific permission and/or a fee. Request permissions from [permissions@acm.org](mailto:permissions@acm.org).

*Haskell '14*, September 4–5, 2014, Gothenburg, Sweden.

Copyright is held by the owner/author(s). Publication rights licensed to ACM.

ACM 978-1-4503-3041-1/14/09...\$15.00.

<http://dx.doi.org/10.1145/2633357.2633358>

are not entirely orthogonal: applying handlers in different orders may give rise to different interactions between effects—perhaps the best known example is that of the two possible interactions between state and non-determinism. The flexibility of ordering handlers is of course crucial: we need control over the interaction of effects to obtain the right semantics for a particular application. However, if handlers double as scoping constructs, the two roles may be at odds: one order of the handlers provides the right scopes and the other order provides the right semantics. Unfortunately, we cannot have it both ways.

This paper solves the dilemma by shifting the responsibility of creating scopes from handlers to syntax. This way we can safely reorder handlers to control the interaction semantics while scoping is unaffected. Of course, handlers are still responsible for assigning a semantics to syntax that create scopes.

The specific contributions of this paper are:

1. We present a gentle and accessible introduction to the effect handlers approach and its infrastructure.
2. We provide several examples that demonstrate the problem of scoping through handlers: pruning nondeterministic choices, exception handling, and multi-threading.
3. We develop two different approaches for handling scoping through syntax.
  - (a) First, we use syntax within the existing effect handlers framework to delimit scopes, and show how to write a handler that works with this syntax. This solution is conceptually lightweight, since it makes use of syntax and is nothing other than another handler. However, it is not general enough to capture

syntax that truly requires programs as arguments.

- (b) As a second solution we provide higher-order syntax that truly allows to embed programs within scoping constructs. This solution is more general, but requires a substantial adaptation of the effect handlers approach.

4. We illustrate both syntax scoping approaches on the examples and show how they effectively solve the problem.

The remainder of this paper is structured as follows. The first part provides background on the effect handlers approach through a number of examples and sets up the necessary infrastructure. We start with a gentle introduction to handling backtracking computations in Section 2. In Section 3 we prepare the ground for more modular syntax by using the datatypes à la carte approach. We demonstrate this modularity in Section 4, where we show how state can be added to nondeterministic computation. We then show how handlers can span different syntax signatures in Section 5.

The second part of this paper focuses on scoped effects. Section 6 builds grammars to parse input, and shows how using handlers to create local scopes imposes undesired semantics. In Section 7 we fix this problem by using syntax to delimit scope. Section 8 demonstrates exception handling as another example that requires

scoped effects, which is resolved in Section 9. We show a more robust solution to the problem in Section 10, where higher-order syntax is introduced. Section 11 gives an example where our first-order approach fails, but that can be solved with higher-order syntax. Finally, we discuss related work in Section 12 and conclude in Section 13.

## 2. Backtracking Computation

The effect handlers approach splits the problem of modelling behaviour into two parts. First syntax is introduced to represent the actions of interest; second, so-called handlers are written that interpret syntax trees into a semantic domain.

For instance, to model the behaviour of backtrackable computation, we use the datatype *Backtr a* to represent the syntax.

```
data Backtr a
  = Return a
  | Fail
  | Backtr a :| Backtr a
```

Here, *Return x* represents a successful computation witnessed by *x*, *Fail* is for computations that have failed, and choice is given by *p :| q*, where *p* and *q* are backtrackable computations.

This representation of backtrackable computations forms a monad, which will allow us to conveniently put syntax together.

```
instance Monad Backtr where
  return a = Return a
  Return a >>= r = r a
  Fail >>= r = Fail
  (p :| q) >>= r = (p >>= r) :| (q >>= r)
```

This monad instance allows us to conveniently write programs that piggyback on Haskell’s **do** notation. As an example, consider how we might solve the well-known knapsack problem, where we choose elements from  $vs$  that sum to  $w$ . Assuming the values are all positive, the following is a naive solution to the problem that goes through all the different possibilities.

$$\begin{aligned} \text{knapsack} &:: \text{Int} \rightarrow [\text{Int}] \rightarrow \text{Backtr} [\text{Int}] \\ \text{knapsack } w \text{ } vs & \mid w < 0 = \text{Fail} \\ & \mid w == 0 = \text{return } [] \\ & \mid w > 0 = \text{do } v \leftarrow \text{select } vs \\ & \quad \quad \quad vs' \leftarrow \text{knapsack } (w - v) \text{ } vs \\ & \quad \quad \quad \text{return } (v : vs') \end{aligned}$$

This makes use of the *select* function that turns a list of values into backtrackable computations:

$$\begin{aligned} \text{select} &:: [a] \rightarrow \text{Backtr } a \\ \text{select} &= \text{foldr } (:[]) \text{ Fail } \cdot \text{map Return} \end{aligned}$$

We fail when there are no values left to select, otherwise, we offer the choice between a given value and the remaining ones.


The resulting construction of *knapsack* 3 [3,2,1] is a tree that expresses the decisions that are made when choosing from the list of values. Figure 1 shows the syntax tree.

We can extract successful computations by making use of the function *allsols*  $p$ , which produces a list of all the solutions that are generated by the program  $p$ . In fact, *allsols* is our first example of a *handler*: it takes the syntax of a backtrackable computation, and handles it to produce a list of all solutions.

$$\begin{aligned} \text{allsols} &:: \text{Backtr } a \rightarrow [a] \\ \text{allsols } (\text{Return } a) &= [a] \end{aligned}$$

$$\begin{aligned} \text{allsols}(\text{Fail}) &= [] \\ \text{allsols}(p : [q]) &= \text{allsols } p ++ \text{allsols } q \end{aligned}$$

2



**Figure 1.** The syntax tree of *knapsack* 3 [3, 2, 1]

Putting the different parts together, we can use *allsols* to capture the solutions to the *knapsack* problem:

$$allsols(knapsack\ 3\ [3,2,1]) = [[3],[2,1],[1,2],[1,1,1]]$$

We characterize *allsols* as a handler since it turns the syntax tree of nondeterministic choices into the semantic domain of solutions.

### 3. Syntax Signatures

We can generalize away from backtrackable computations by defining a datatype that is parametric in the signature of the syntax. We

factor syntax for programs into the *Return* constructor and a constructor *Op* that allows us to incorporate operations of interest from some signature *sig*.

```
data Prog sig a
  = Return a                -- pure computations
  | Op (sig (Prog sig a))   -- impure computations
```

For instance, the operations that give rise to computations of type *Backtr a* are captured by the signature functor *Nondet*, such that  $\text{Backtr } a \cong \text{Prog Nondet } a$ .

```
data Nondet cnt
  = Fail'
  | cnt : []' cnt
```

The type argument *cnt* marks the recursive components, which in this context are continuations into some other syntactic construct. Here, and elsewhere in the paper, we assume that the functor instance is automatically derived.

This abstraction has bought us an important benefit: the *Prog sig* type forms a monad whenever *sig* is a functor (it is the free monad for the functor *sig*), which helps us to easily compose programs together from constituent parts.

```
instance (Functor sig)  $\Rightarrow$  Monad (Prog sig) where
  return v = Return v

  Return v  $\gg\!=\!$  prog = prog v
  Op op     $\gg\!=\!$  prog = Op (fmap (>>= prog) op)
```

In the monad instance, we can read  $(\gg\!=\!)$  as substitution, where each fragment of syntax represents a first-order term with variables:



*Return*  $v$  is a variable, and *Op op* is a compound term. Another reading is that *Prog sig a* represents computations or programs: *Return*  $v$  is a pure computation, *Op op* is an impure computation, and here ( $\gg$ ) represents the sequential chaining of computations.

**Syntax Infrastructure** So far we have looked at only one syntax signature. More generally, we will be dealing with several different effects that work together in a single program, and so we need a means of flexibly composing signatures, where each signature captures syntax that encodes a particular effect.

Perhaps the simplest way to compose signatures is with the coproduct, where two signatures  $sig_1$  and  $sig_2$  are combined.

**data**  $(sig_1 + sig_2) \text{ cnt} = \text{Inl } (sig_1 \text{ cnt}) \mid \text{Inr } (sig_2 \text{ cnt})$

Handlers over such signatures must be run one after the other, each dealing with part of the signature they recognize. In practice, this becomes a little cumbersome since the handlers have to carefully invoke the right mixture of *Inl* and *Inr* constructors to get access to the syntax they are interested in.

To fix this, we want to be able to inject and project constructors for some syntax into some larger language, and this is where the *datatypes à la carte* technique [14] shines, by precisely expressing the relationship between families of syntax:

```
class (Functor sub, Functor sup)  $\Rightarrow$  sub  $\subset$  sup where
  inj :: sub a  $\rightarrow$  sup a
  prj :: sup a  $\rightarrow$  Maybe (sub a)
instance Functor sig  $\Rightarrow$  sig  $\subset$  sig where
  inj = id
  prj = Just
```

The coproduct fits into this scheme nicely, as is evidenced by the following instances:

```
instance (Functor sig1, Functor sig2)  $\Rightarrow$ 
  sig1  $\subset$  (sig1 + sig2) where
  inj           = Inl
  prj (Inl fa) = Just fa
  prj _         = Nothing
instance (Functor sig1, sig  $\subset$  sig2)  $\Rightarrow$ 
  sig  $\subset$  (sig1 + sig2) where
  inj           = Inr · inj
  prj (Inr ga) = prj ga
  prj _         = Nothing
```

This gives a nice way of composing signatures, and makes it possible

to inject syntax into programs over those signatures.

$$\begin{aligned} \text{inject} &:: (\text{sub} \subset \text{sup}) \Rightarrow \text{sub} (\text{Prog sup } a) \rightarrow \text{Prog sup } a \\ \text{inject} &= \text{Op} \cdot \text{inj} \end{aligned}$$

As well as conveniently building operations of a program, we have a way of extracting operations from that program for inspection.

$$\begin{aligned} \text{project} &:: (\text{sub} \subset \text{sup}) \Rightarrow \text{Prog sup } a \rightarrow \text{Maybe} (\text{sub} (\text{Prog sup } a)) \\ \text{project} (\text{Op } s) &= \text{prj } s \\ \text{project } \_ &= \text{Nothing} \end{aligned}$$

This projection returns the syntax of interest when we are dealing with an *Op*, and otherwise returns *Nothing*.

With smart constructors and destructors we can make the embedding of syntax into a wider context relatively painless. We make use of *pattern synonyms* and *view patterns*, which are recently implemented extensions in GHC<sup>1</sup>.

This allows us to create a new pattern called *Fail*, which works on programs of type *Prog sig a* for any signature where *Nondet*  $\subset$  *sig*. In the following pattern declaration, the function *project* is applied at the site of the pattern, and if the result is *Just Fail'*, the match is a success.

$$\begin{aligned} \text{pattern Fail} &\leftarrow (\text{project} \rightarrow \text{Just Fail}') \\ \text{fail} &:: (\text{Nondet} \subset \text{sig}) \Rightarrow \text{Prog sig } a \\ \text{fail} &= \text{inject Fail}' \end{aligned}$$

For choice we can again use the same approach:

$$\begin{aligned} \text{pattern } p \llbracket q &\leftarrow (\text{project} \rightarrow \text{Just } (p \llbracket' q)) \\ (\llbracket) &:: (\text{Nondet} \subset \text{sig}) \Rightarrow \text{Prog sig } a \rightarrow \text{Prog sig } a \rightarrow \text{Prog sig } a \\ p \llbracket q &= \text{inject } (p \llbracket' q) \end{aligned}$$

---

<sup>1</sup>[http://www.haskell.org/ghc/docs/7.8.2/users\\_guide.pdf](http://www.haskell.org/ghc/docs/7.8.2/users_guide.pdf)

This time applying *project* to a program might yield  $Just (p : \llbracket' q)$ , and if this is the case we bind the variables in that context to those in the pattern  $p : \llbracket q$ .

We will be assembling programs using coproducts. Since our goal is to inject syntax into some accumulating collection of syntax functors we need a base case, and this is provided by *Void*, which is the signature for empty syntax:

**data** *Void cnt*

Using this syntax functor as a signature results in programs of type *Prog Void a*, where it is impossible to use the *Op* constructor. We can nevertheless extract values from such programs:

$$\begin{aligned} run &:: Prog\ Void\ a \rightarrow a \\ run\ (Return\ x) &= x \end{aligned}$$

This handler is usually the last one to be run, since it extracts a final value from a program with no more syntax.

When providing semantics for programs with signatures made up of coproducts, we will define handlers that deal with a specific part of that signature, and leaves the rest untouched. This is the key to modular semantics, allowing us to focus on the interesting details. We use *Other* to represent the other syntax that is not interesting in a given context:

**pattern**  $Other\ s = Op\ (Inr\ s)$

For instance, we can evaluate a program with *Nondet* syntax on the left of its signature by using the *solutions* function, which is highly modular since *sig* can be an arbitrary signature:

$$\begin{aligned} solutions &:: (Functor\ sig) \Rightarrow Prog\ (Nondet + sig)\ a \rightarrow Prog\ sig\ [a] \\ solutions\ (Return\ a) &= return\ [a] \\ solutions\ (Fail) &= return\ [] \\ solutions\ (p \mathbin{::} q) &= liftM2\ (++)\ (solutions\ p)\ (solutions\ q) \\ solutions\ (Other\ op) &= Op\ (fmap\ solutions\ op) \end{aligned}$$

This is a lifted, or monadized, version of *allsols* where there might be syntax other than that given by *Nondet* involved. We can recover *allsols* by noticing that  $Backtr\ a \cong Prog\ (Nondet + Void)\ a$ , and adapting the definition to fit our more modular framework:

$$\begin{aligned} allsols &:: Prog\ (Nondet + Void)\ a \rightarrow [a] \\ allsols &= run \cdot solutions \end{aligned}$$

This approach uses *run* to extract results from a program that has no more syntax.

## 4. Composing Semantics

The main point of the effect handlers approach is that both the syntax *and* the semantics of different effects can be trivially composed. In other words, effect handlers provide modular semantics. We illustrate this point by combining nondeterminism with state.

### 4.1 The State Effect

Stateful operations are modelled with the assumption that there exists some underlying state *s*, which can be updated with the

operation *put s*, and retrieved with *get*. The corresponding syntax is:

**data** *State s cnt*  
= *Get'* (*s* → *cnt*)  
| *Put'* *s cnt*

**pattern** *Get k* ← (*project* → *Just* (*Get'* *k*))  
*get* :: (*State s* ⊂ *sig*) ⇒ *Prog sig s*  
*get* = *inject* (*Get'* *return*)

**pattern** *Put s k* ← (*project* → *Just* (*Put'* *s k*))  
*put* :: (*State s* ⊂ *sig*) ⇒ *s* → *Prog sig* ()  
*put s* = *inject* (*Put'* *s* (*return* ()))

If we want to execute a stateful computation, then we can use the following handler, which takes an initial state, and a program that contains state manipulating syntax to return a residual program which returns an output state.

$$\begin{aligned}
\text{runState} &:: \text{Functor } \text{sig} \Rightarrow \\
& s \rightarrow \text{Prog } (\text{State } s + \text{sig}) \ a \rightarrow \text{Prog } \text{sig } (s, a) \\
\text{runState } s \ (\text{Return } a) &= \text{return } (s, a) \\
\text{runState } s \ (\text{Get } k) &= \text{runState } s \ (k \ s) \\
\text{runState } s \ (\text{Put } s' \ k) &= \text{runState } s' \ k \\
\text{runState } s \ (\text{Other } op) &= \text{Op } (fmap \ (\text{runState } s) \ op)
\end{aligned}$$

This works by carrying around the appropriate state in recursive calls: when a new state is inserted with *Put*  $s'$ , then this new state  $s'$  replaces the previous one.

## 4.2 Combining State and Nondeterminism

Now we can assign a semantics to syntactic programs that combine nondeterminism and state by providing the semantics for both effects separately: we just compose both handlers. The first handler tackles one effect in the initial program while the second handler tackles the other in the residual program.

It is vital to note that we have a degree of freedom when composing two handlers: we can choose which handler to apply first. For instance, for the *runState* and *solutions* handlers we can choose between either *runLocal* or *runGlobal*:

$$\begin{aligned}
\text{runLocal} &:: \text{Functor } \text{sig} \Rightarrow \\
& s \rightarrow \text{Prog } (\text{State } s + \text{Nondet} + \text{sig}) \ a \rightarrow \text{Prog } \text{sig } [(s, a)] \\
\text{runLocal } s &= \text{solutions} \cdot \text{runState } s \\
\text{runGlobal} &:: \text{Functor } \text{sig} \Rightarrow \\
& s \rightarrow \text{Prog } (\text{Nondet} + \text{State } s + \text{sig}) \ a \rightarrow \text{Prog } \text{sig } (s, [a])
\end{aligned}$$

$$\text{runGlobal } s = \text{runState } s \cdot \text{solutions}$$

These two composite semantics are not equivalent; they differ in how the two effects interact. Here we get two flavors of nondeterministic state: *local* and *global*. In *runLocal*, each branch of the nondeterministic computation has its own local copy of the state, while in *runGlobal* there is one state shared by all branches. The difference between the two is also apparent at the type level: *runLocal* returns a list of different final values with their associated states, one state for each solution in the backtracking. The type of *runGlobal* reveals that it produces a list of alternative solutions and only one final state.

The fact that we get different semantics through different compositions is a great benefit of the effect handlers approach: in return for writing the handlers in modular style, we get multiple interaction semantics for free!

The following example illustrates that both flavors of nondeterministic state are useful for different purposes. In fact, the example even shows the use of two different semantics for the same program and involves a third handler that counts how many number choices are made.

$$\begin{aligned} \text{choices} &:: (\text{Nondet} \subset \text{sig}, \text{State } \text{Int} \subset \text{sig}) \\ &\Rightarrow \text{Prog sig } a \rightarrow \text{Prog sig } a \\ \text{choices } (\text{Return } a) &= \text{return } a \\ \text{choices } (\text{Fail}) &= \text{fail} \\ \text{choices } (p \parallel q) &= \text{incr} \gg (\text{choices } p \parallel \text{choices } q) \\ \text{choices } (\text{Op } op) &= \text{Op } (\text{fmap choices } op) \\ \text{incr} &:: (\text{State } \text{Int} \subset \text{sig}) \Rightarrow \text{Prog sig } () \\ \text{incr} &= \text{get} \gg \text{put} \cdot (\text{succ} :: \text{Int} \rightarrow \text{Int}) \end{aligned}$$

The counting is performed by *incr*, which simply increments an *Int* stored in the state. We evaluate an *incr* every time we encounter a



choice, and then recursively count choices in each branch.

In order to apply the *choices* handler to the *knapsack* example, we need to adapt the definitions of both *knapsack* and *select* to the

4

modular setting. Thankfully, this involves only a little more than providing slightly different type signatures; the code body remains unchanged, except that we use the smart constructors *fail* and (`[]`):

$$\begin{aligned} \text{knapsack} &:: (\text{Nondet} \subset \text{sig}) \Rightarrow \text{Int} \rightarrow [\text{Int}] \rightarrow \text{Prog sig} [\text{Int}] \\ \text{select} &:: (\text{Nondet} \subset \text{sig}) \Rightarrow [a] \rightarrow \text{Prog sig } a \end{aligned}$$

Now we can observe that both global and local state give us different information. The global version tells us how many choice points are explored to find all solutions:

```
> (run · runGlobal (0 :: Int) · choices) (knapsack 3 [3,2,1])
(12, [[3], [2,1], [1,2], [1,1,1]])
```

In contrast, the local version tells us exactly how deep in the tree of choices each individual answer is found:

```
> (run · runLocal (0 :: Int) · choices) (knapsack 3 [3,2,1])
[(1,[3]), (5,[2,1]), (5,[1,2]), (9,[1,1,1])]
```

The information provided by the global version cannot be reconstructed from the local information, and vice versa.

**Summary** Composing the semantics of two orthogonal features,

nondeterminism and state, is easy with effect handlers. In fact, we can express the two interactions of state and nondeterminism simply by composing their handlers in different orders.

## 5. Cut and Call

The effect handlers approach does not force us to write orthogonal handlers. This section shows that we can extend nondeterminism with a non-orthogonal feature. The *Cutfail* operation immediately ends the search with failure, dropping all extant unexplored branches. Hence, there is clearly interaction with *Nondet*.

```
data Cut cnt = Cutfail'
pattern Cutfail ← (project → Just Cutfail')
cutfail :: (Cut ⊆ sig) ⇒ Prog sig a
cutfail = inject Cutfail'
```

The expression *call p*, defined in terms of the *go* handler, delimits the action of *Cutfail* in a program *p*.

```
call :: (Nondet ⊆ sig) ⇒ Prog (Cut + sig) a → Prog sig a
call p = go p fail where
  go :: (Nondet ⊆ sig) ⇒
    Prog (Cut + sig) a → Prog sig a → Prog sig a
  go (Return a) q = return a ∥ q
  go (Fail) q = q
  go (Cutfail) q = fail
  go (p1 ∥ p2) q = go p1 (go p2 q)
  go (Other op) q = Op (fmap (flip go q) op)
```

The *go p q* handler accumulates in its second parameter *q* the

unexplored alternatives to  $p$ . When  $go$  encounters a *Return* or a *Fail*, it explores the alternatives in  $q$ . When a *Cutfail* is encountered, the computation fails immediately, without exploring any alternatives. At a branching,  $go$  explores the left branch and adds the right branch to the unexplored alternatives.

Often *cutfail* is used in the form of *cut*, which can be defined as:

$$cut :: (Nondet \subset sig, Cut \subset sig) \Rightarrow Prog\ sig\ ()$$
$$cut = skip \parallel cutfail$$
$$skip :: Monad\ m \Rightarrow m\ ()$$
$$skip = return\ ()$$

This commits the computation to the current branch, pruning any unexplored alternatives. For example, *once*  $p$  commits to the first solution that is found in  $p$ .

$$\begin{aligned} \text{once} &:: (\text{Nondet} \subset \text{sig}) \Rightarrow \text{Prog} (\text{Cut} + \text{sig}) b \rightarrow \text{Prog sig } b \\ \text{once } p &= \text{call } (\mathbf{do} \ x \leftarrow p; \text{cut}; \text{return } x) \end{aligned}$$

This way we can compute only the first *knapsack* solution as follows.

$$\begin{aligned} &> (\text{run} \cdot \text{solutions} \cdot \text{once}) (\text{knapsack } 3 \ [3, 2, 1]) \\ &[[3]] \end{aligned}$$

In summary, we can write non-orthogonal handlers like *call* just as easily as modular ones like *solutions* and they play nicely together. However, there lurks a deep problem in these murky waters, where *call* does not always behave quite as we expect it to. We explore this in the next section with a different example.

## 6. Grammars

This section establishes the central problem tackled in this paper. We call a handler like *call* a *scoping* handler, because it not only provides the semantics for particular syntax, but also creates a local scope in which the impact of an effect is contained. The two roles of scoping handlers can be fundamentally at odds with one another: different orders of handlers affect the interaction semantics, while different scopes affect the extent of an effect's impact. With scoping handlers these two choices are not independent; we cannot affect one without the other. Yet, often we need to control both separately. This section illustrates that point on grammars.

**Grammars** Grammars can be expressed compactly using syntax signatures from a remarkably small base: the *Symbol* functor represents syntax that matches a single symbol from some source of characters:

$$\begin{aligned} \mathbf{data} \ \text{Symbol } cnt &= \text{Symbol}' \ \text{Char} \ (\text{Char} \rightarrow cnt) \\ \mathbf{pattern} \ \text{Symbol } c \ k &\leftarrow (\text{project} \rightarrow \text{Just } (\text{Symbol}' \ c \ k)) \end{aligned}$$

$$\begin{aligned} \text{symbol} &:: (\text{Symbol} \subset \text{sig}) \Rightarrow \text{Char} \rightarrow \text{Prog sig Char} \\ \text{symbol } c &= \text{inject } (\text{Symbol}' c \text{ return}) \end{aligned}$$

The constructor *symbol c* attempts to match *c* with the current input, and if it succeeds, passes the value of *c* on to its continuation. For instance, we can build a digit recognizer with the following:

$$\begin{aligned} \text{digit} &:: (\text{Nondet} \subset \text{sig}, \text{Symbol} \subset \text{sig}) \Rightarrow \text{Prog sig Char} \\ \text{digit} &= \text{foldr } ([\ ] \text{ fail } (\text{fmap symbol } [\text{'0'} \dots \text{'9'}])) \end{aligned}$$

This nondeterministically attempts to match all of the digits, and fails if this is not possible.

The combinators *many* and *many<sub>1</sub>* will be familiar to readers who have worked with grammar libraries. Their definitions encode an accumulation of values from nondeterministic programs:

$$\begin{aligned} \text{many}, \text{many}_1 &:: (\text{Nondet} \subset \text{sig}) \Rightarrow \text{Prog sig } a \rightarrow \text{Prog sig } [a] \\ \text{many } p &= \text{many}_1 p \text{ } [\ ] \text{ return } [\ ] \\ \text{many}_1 p &= \text{do } a \leftarrow p; \text{ as } \leftarrow \text{many } p; \text{ return } (a : \text{as}) \end{aligned}$$

Both of these functions build nondeterminism into the output of a program that supports *Nondet* syntax.

The *parse xs* handler takes a grammar to a nondeterministic program. It resolves the *Symbol c k* constructors by matching *c* against the first element in the list of characters *xs*, turning it into failure when the match fails, or passing *c* on to the continuation *k* if it succeeds.

$$\begin{aligned} \text{parse} &:: (\text{Nondet} \subset \text{sig}) \Rightarrow \\ &[\text{Char}] \rightarrow \text{Prog } (\text{Symbol} + \text{sig}) a \rightarrow \text{Prog sig } a \\ \text{parse } [\ ] &(\text{Return } a) = \text{return } a \\ \text{parse } (x : \text{xs}) &(\text{Return } a) = \text{fail} \\ \text{parse } [\ ] &(\text{Symbol } c k) = \text{fail} \\ \text{parse } (x : \text{xs}) &(\text{Symbol } c k) \mid x == c = \text{parse } \text{xs } (k x) \end{aligned}$$

$\text{parse } xs \quad (Other \text{ op}) \quad | \text{ otherwise} = fail$   
 $= Op \text{ (fmap (parse } xs) \text{ op)}$

This handler also fails if the input is not entirely consumed, or if the grammar expects more symbols.

Parsing arithmetic expressions which are made up of sums and products can be done using the traditional recipe, where an *expr* deals with sums, and a *term* deals with products. We return the result of evaluating the payload directly.

```

expr :: (Nondet ⊂ sig, Symbol ⊂ sig) ⇒ Prog sig Int
expr = do i ← term; symbol '+'; j ← expr; return (i + j)
      | do i ← term; return i
term :: (Nondet ⊂ sig, Symbol ⊂ sig) ⇒ Prog sig Int
term = do i ← factor; symbol '*'; j ← term; return (i * j)
      | do i ← factor; return i

```

The terminal case is in *factor*, which is either a string of digits, or an expression in parentheses:

```

factor :: (Nondet ⊂ sig, Symbol ⊂ sig) ⇒ Prog sig Int
factor = do ds ← many1 digit; return (read ds)
      | do symbol '('; i ← expr; symbol ')'; return i

```

To parse an expression, we simply handle the program with *parse*:

```

> (allsols · parse "2+8*5") expr
[42]

```

**Grammar Refactoring** We can left factor our grammars *expr* and *term* to improve efficiency. Focusing on *expr*, we factor out the common *term* prefix in the two branches.

$$\begin{aligned} \text{expr}_1 &:: (\text{Nondet} \subset \text{sig}, \text{Symbol} \subset \text{sig}) \Rightarrow \text{Prog sig Int} \\ \text{expr}_1 &= \mathbf{do} \ i \leftarrow \text{term} \\ &\quad ( \mathbf{do} \ \text{symbol } '++' ; j \leftarrow \text{expr}_1 ; \text{return } (i+j) \\ &\quad \quad \parallel \mathbf{do} \ \text{return } i ) \end{aligned}$$

In the refactored *expr*<sub>1</sub> grammar, the two branches are mutually exclusive. The reason is that the first branch requires the next character in the input to be a '++', while the second branch can only be followed by a ')' or the end of the input. Hence, after seeing a '++' we can safely commit to the first branch and prune the second branch. Pruning the alternative should have a beneficial effect on performance because the parser will no longer unnecessarily explore the alternative.

In the previous section we introduced the control operator *cut* to commit to a successful branch. In this case, we want to commit to the first branch when a *symbol* '++' is encountered, so we might try the following:

$$\begin{aligned} \text{expr}_2 &:: (\text{Nondet} \subset \text{sig}, \text{Symbol} \subset \text{sig}) \Rightarrow \text{Prog sig Int} \\ \text{expr}_2 &= \mathbf{do} \ i \leftarrow \text{term} \\ &\quad \text{call } ( \mathbf{do} \ \text{symbol } '++' ; \text{cut} ; j \leftarrow \text{expr}_2 ; \text{return } (i+j) \\ &\quad \quad \parallel \mathbf{do} \ \text{return } i ) \end{aligned}$$

At a first glance, this *seems* sensible: the locally placed *call* handler is needed to delimit the scope of the *cut*. After all, the *cut* is only meant to prune the one alternative, and should not affect other alternatives made elsewhere in the grammar.

**The Problem** Alas, while the above grammar syntactically captures the desired pruning, it may come as a surprise that the handlers do not provide the desired semantics:

```
> (allsols · parse "1") expr2  
[]
```

We expect the result `[1]`, but the parse fails instead. In order to understand why this happens, we need to carefully consider the impact of the following clause in the definition of the subsidiary *go* handler of *call*.

$$go\ (Other\ op)\ q = Op\ (fmap\ (flip\ go\ q)\ op)$$



After one recursive invocation of *call* in *expr*<sub>2</sub>, this clause matches the “other” operation *symbol* ‘+’. In effect, we can think of this execution as rewriting the body of *expr*<sub>2</sub> to:

```
do i ← term  
    symbol ‘+’  
    go (do cut; j ← expr2; return (i + j))  
    (return i ∥ fail)
```

In other words, *symbol* ‘+’ has been hoisted out of the left branch and now happens before the *call*. Hence, the input always has to contain a ‘+’; this is obviously not what we want.

**A Non-Solution** The problem is that we have chosen the wrong order for the *parse* and *call* handlers, which leads to the undesired interaction. The appropriate interaction is obtained by first applying *parse* and then *call*. This way there is no more *symbol* for *call* to hoist out of a branch.

Unfortunately, we cannot reorder *call* and *parse* for other reasons: *call* creates a local scope. We cannot put it anywhere else without risking that *cut* prunes more alternatives than it should. Conversely, it obviously does not make sense to apply *parse* only in the local scope of *call*. Hence, we are stuck and *call* is to blame because it unnecessarily couples scoping and semantics.

**Summary** This section has shown that the coupling of scoping and semantics in scoping handlers is problematic. In the following sections we look at two different solutions to this problem; both solutions decouple scoping from semantics by making scoping the province of syntax. The first solution uses lightweight syntax that fits naturally into the first-order framework of effect handlers we have been describing so far, but is prone to user errors. The second solution we look at is more robust and expressive, but requires much

heavier machinery.

## 7. Scoped Syntax

One solution to the problem we encounter with *call* is to explicitly delimit the beginning and end of the scope of the *call*. This can be managed by giving the user syntax to explicitly set these boundaries:

```
data Call cnt = BCall' cnt | ECall' cnt  
pattern BCall p  $\leftarrow$  (project  $\rightarrow$  Just (BCall' p))  
pattern ECall p  $\leftarrow$  (project  $\rightarrow$  Just (ECall' p))
```

We will want to ensure that each *BCall* is paired with an *ECall*. Hence, they should only be exposed to the user in the form of *call'*.

```
call' :: (Call  $\subset$  sig)  $\Rightarrow$  Prog sig a  $\rightarrow$  Prog sig a  
call' p = do begin; x  $\leftarrow$  p; end; return x where  
  begin = inject (BCall' (return ()))  
  end   = inject (ECall' (return ()))
```

With these changes, the left factored expression remains syntactically the same as *expr*<sub>2</sub>, except that *call* has been replaced by *call'*, and the signature is now more elaborate, since the program explicitly incorporates *Cut* and *Call* syntax.

```
expr3 :: (Nondet  $\subset$  sig, Symbol  $\subset$  sig, Call  $\subset$  sig, Cut  $\subset$  sig)  $\Rightarrow$   
  Prog sig Int  
expr3 = do i  $\leftarrow$  term  
          call' ( do symbol ' + ' ; cut; j  $\leftarrow$  expr3; return (i + j)  
                do return i)
```

We can run this with the *runCut* handler, that provides semantics to *cut* in a way that respects the scope set out by *BCall* and *ECall*.

```
> run · solutions · runCut · parse "1" $expr3  
[1]
```

6

While on the surface not much has changed, there is a lot more going on behind the scenes.

The *runCut* handler is now used to eliminate *Call* and *Cut* from the signature:

$$\begin{aligned} \text{runCut} &:: (\text{Nondet} \subset \text{sig}) \Rightarrow \\ &\text{Prog} (\text{Call} + \text{Cut} + \text{sig}) a \rightarrow \text{Prog sig } a \\ \text{runCut } p &= \text{call } (\text{bcall } p) \end{aligned}$$

The definition of *runCut* is in terms of two helper functions, *bcall* and *ecall*. The interesting case for *bcall* is when it encounters a *BCall* *p*. In this case, the handler *call* is used to handle the code in the continuation *p* up until the matching *ecall* *q*, which is found by the function *ecall*.

$$\begin{aligned} \text{bcall} &:: (\text{Nondet} \subset \text{sig}) \Rightarrow \\ &\text{Prog} (\text{Call} + \text{Cut} + \text{sig}) a \rightarrow \text{Prog} (\text{Cut} + \text{sig}) a \\ \text{bcall} (\text{Return } a) &= \text{return } a \\ \text{bcall} (\text{BCall } p) &= \text{upcast } (\text{call } (\text{ecall } p)) \gg\gg \text{bcall} \\ \text{bcall} (\text{ECall } p) &= \text{error "Mismatched ECall!"} \\ \text{bcall} (\text{Other } op) &= \text{Op } (\text{fmap } \text{bcall } op) \end{aligned}$$

If an *ECall* is found during the execution of *begin*, then an error is raised, since this must be a mismatched *ECall* *p*. An alternative to raising an error is to simply ignore the spurious *ECall*, and continue with *p*.

The function *ecall* takes a program with scoped syntax and modifies it so that any scope context is removed. The code outside of that scope is found in *ECall p*, where  $p :: \text{Prog } (\text{Call} + \text{Cut} + \text{sig})$  *a* is a program in its own right.

$$\begin{aligned} \text{ecall} &:: (\text{Nondet} \subset \text{sig}) \Rightarrow \\ &\quad \text{Prog } (\text{Call} + \text{Cut} + \text{sig}) \text{ } a \rightarrow \\ &\quad \text{Prog } (\text{Cut} + \text{sig}) (\text{Prog } (\text{Call} + \text{Cut} + \text{sig}) \text{ } a) \\ \text{ecall } (\text{Return } a) &= \text{return } (\text{Return } a) \\ \text{ecall } (\text{BCall } p) &= \text{upcast } (\text{call } (\text{ecall } p)) \gg \text{ecall} \\ \text{ecall } (\text{ECall } p) &= \text{return } p \\ \text{ecall } (\text{Other } op) &= \text{Op } (\text{fmap } \text{ecall } op) \end{aligned}$$

Since *call* removes all the syntax given by *Cut* from the signature, we use *upcast* to ensure that the types match our expectations. The function *upcast* simply extends a signature so that it contains an additional syntax functor. It works by shifting operations in the original signature into the right of the resulting coproduct.

$$\begin{aligned} \text{upcast} &:: (\text{Functor } f, \text{Functor } \text{sig}) \Rightarrow \text{Prog } \text{sig } a \rightarrow \text{Prog } (f + \text{sig}) \text{ } a \\ \text{upcast } (\text{Return } x) &= \text{return } x \\ \text{upcast } (\text{Op } op) &= \text{Op } (\text{Inr } (\text{fmap } \text{upcast } op)) \end{aligned}$$

In summary, the idea of the tagging technique is to mark the beginning and the end of a scope with syntactic operations, the tags. Handlers should take these tags into account to determine the impact of effects.

While this tagging of scope seems like a neat solution, we find it lacking in several regards. For one, we require the user to carefully ensure that scopes are nested correctly with *begin* and *end*. We have solved this with some syntactic sugar, but this could be circumvented unless the constructors are removed from the

programmer's vocabulary. This can be achieved by abstraction.

Another criticism is that this solution is perhaps not as general as we would hope. For instance, it is insufficient to solve the related problem of expressing the scope of code with exception handling, which we explore more carefully in the following section.

## **8. Exceptions**

This section presents a second instance of the scoping handler problem in the form of exception handling. Exception handling is a fundamental feature of many programming languages. It allows a

block of code to terminate abruptly in a way that throws an exception value from which the overall program resumes computation. We will model this with effect handlers.

The syntax for exceptions is as follows, where an exception value of type  $e$  is thrown by the syntax  $Throw^! e$ :

**data**  $Exc\ e\ cnt = Throw^! e$   
**pattern**  $Throw\ e \leftarrow (project \rightarrow Just\ (Throw^! e))$   
 $throw :: (Exc\ e \subset sig) \Rightarrow e \rightarrow Prog\ sig\ a$   
 $throw\ e = inject\ (Throw^! e)$

To handle a thrown exception, we use  $runExc$ :

$runExc :: Functor\ sig \Rightarrow$   
 $Prog\ (Exc\ e + sig)\ a \rightarrow Prog\ sig\ (Either\ e\ a)$   
 $runExc\ (Return\ x) = return\ (Right\ x)$   
 $runExc\ (Throw\ e) = return\ (Left\ e)$   
 $runExc\ (Other\ op) = Op\ (fmap\ runExc\ op)$

This handler uses the standard approach of encoding exceptions into values of type  $Either\ e\ a$ : computations normally place their results in  $Right\ a$ , unless an exception  $e$  is thrown, in which case this is signalled with the value  $Left\ e$ .

In addition to propagating exceptions into the outer program, we are also interested in catching exceptions in code, and handling them with some computation that can recover. This can be modelled by the following handler, where  $catch\ p\ h \gg\!\!= k$  executes  $p$ , and continues with  $k$  unless an exception  $e$  is thrown, in which case  $h\ e \gg\!\!= k$  is invoked.

$catch :: (Exc\ e \subset sig)$   
 $\Rightarrow Prog\ sig\ a \rightarrow (e \rightarrow Prog\ sig\ a) \rightarrow Prog\ sig\ a$   
 $catch\ (Return\ x)\ h = return\ x$

$$\begin{aligned} \text{catch } (\text{Throw } e) \ h &= h \ e \\ \text{catch } (\text{Op } op) \ h &= \text{Op } (fmap (\lambda p \rightarrow \text{catch } p \ h) \ op) \end{aligned}$$

While this handler is perfectly reasonable at first glance, it suffers from the same problem as our initial version of *call*: it does not compose as flexibly as it could.

As a simple example, consider the interaction of exceptions and state. The following code attempts to decrement the state counter three times, and if an exception is thrown it is handled with *return*:

$$\begin{aligned} \text{tripleDecr} &:: (\text{State } \text{Int} \subset \text{sig}, \text{Exc } () \subset \text{sig}) \Rightarrow \text{Prog } \text{sig } () \\ \text{tripleDecr} &= \text{decr} \gg \text{catch } (\text{decr} \gg \text{decr}) \ \text{return} \\ \text{decr} &:: (\text{State } \text{Int} \subset \text{sig}, \text{Exc } () \subset \text{sig}) \Rightarrow \text{Prog } \text{sig } () \\ \text{decr} &= \mathbf{do} \ x \leftarrow \text{get} \\ &\quad \mathbf{if} \ x > (0 :: \text{Int}) \ \mathbf{then} \ \text{put } (\text{pred } x) \ \mathbf{else} \ \text{throw } () \end{aligned}$$

The *decr* decrements a counter held in state. However, if the decrement would result in a negative value, then the state is left unchanged and an exception is thrown.

If we run *tripleDecr* on a state that initially contains 2, then an exception will be raised by the third *decr*. In this scenario, there are two different reasonable final states to expect: a global interpretation would result in a final state of 0, where the first two *decrs* persist; a more local interpretation would lead us to expect all of the effects within the *catch* to be rolled back, so that the final state 1 is the result of the first *decr* only.

Obtaining different behaviours should in principle be possible by reordering handlers. However, since *catch* is a scoping handler that creates a local scope, we can only express the global interpretation.

$$\begin{aligned} &> (\text{run} \cdot \text{runExc} \cdot \text{runState } 2) \ \text{tripleDecr} \\ &\text{Right } (0, ()) \end{aligned}$$

The *catch* embedded within *tripleDecr* cannot be exchanged with *runState*, because it would change the scope created by *catch*.

7

## 9. Scoped Syntax Revisited

We already noted that the scoped syntax in Section 7 is insufficient to capture the behaviour of exceptions. The issue is that a *catch* block has two different continuations in addition to the body that is to be executed: one continuation in the case where no exceptions are thrown, and another for the exception handler.

We can solve the problem by extending the idea of using tags to delineate the different blocks of code involved.

```
data Catch e cnt = BCatch' cnt (e → cnt) | ECatch' cnt
pattern BCatch p q ← (project → Just (BCatch' p q))
pattern ECatch p ← (project → Just (ECatch' p))
```

Instead of exposing smart constructors *bcatch* and *ecatch*, we instead introduce the following syntactic sugar that ensures the tags are matched appropriately:

```
catch' :: ∀ sig e a . (Catch e ⊂ sig) ⇒
  Prog sig a → (e → Prog sig a) → Prog sig a
catch' p h = begin (do x ← p; end; return x) h where
  begin p q = inject (BCatch' p q)
  end = inject (ECatch' (return ())) :: Catch e (Prog sig ()))
```

Notice that the constructor for *bcatch* does not make use of *return*,



which is what we have done in every continuation parameter so far: instead, the syntax *bcatch p q* takes two continuations explicitly, where *p* represents the code that is to be tried, and *q* is the code that handles potential exceptions.

The *runCatch* function is similar to *runCut*: it handles exceptions, and is defined in terms of the functions *bcatch* and *ecatch*.

$$\begin{aligned} \text{runCatch} &:: (\text{Functor } \text{sig}) \Rightarrow \\ &\quad \text{Prog } (\text{Catch } e + (\text{Exc } e + \text{sig})) \ a \rightarrow \text{Prog } \text{sig } (\text{Either } e \ a) \\ \text{runCatch } p &= \text{runExc } (\text{bcatch } p) \end{aligned}$$

The function *bcatch* searches for a *BCatch p q*, and when one is encountered, it recursively runs exception handling on *p*. If an exception is raised, then the handling code *q* is used instead.

$$\begin{aligned} \text{bcatch} &:: (\text{Functor } \text{sig}) \Rightarrow \\ &\quad \text{Prog } (\text{Catch } e + (\text{Exc } e + \text{sig})) \ a \rightarrow \text{Prog } (\text{Exc } e + \text{sig}) \ a \\ \text{bcatch } (\text{Return } a) &= \text{return } a \\ \text{bcatch } (\text{BCatch } p \ q) &= \\ &\quad \mathbf{do} \ r \leftarrow \text{upcast } (\text{runExc } (\text{ecatch } p)) \\ &\quad (\text{bcatch } \cdot \text{either } q \ \text{id}) \ r \\ \text{bcatch } (\text{ECatch } p) &= \text{error "Mismatched ECatch!"} \\ \text{bcatch } (\text{Other } op) &= \text{Op } (\text{fmap } \text{bcatch } op) \end{aligned}$$

The scope is delimited by an *ECatch*, which is handled by *ecatch*. This implementation mirrors that of *ecall*.

$$\begin{aligned} \text{ecatch} &:: (\text{Functor } \text{sig}) \Rightarrow \\ &\quad \text{Prog } (\text{Catch } e + (\text{Exc } e + \text{sig})) \ a \rightarrow \\ &\quad \text{Prog } (\text{Exc } e + \text{sig}) \ (\text{Prog } (\text{Catch } e + (\text{Exc } e + \text{sig})) \ a) \\ \text{ecatch } (\text{Return } a) &= \text{return } (\text{Return } a) \\ \text{ecatch } (\text{BCatch } p \ q) &= \mathbf{do} \ r \leftarrow \text{upcast } (\text{runExc } (\text{ecatch } p)) \end{aligned}$$

$$(ecatch \cdot either \ q \ id) \ r$$

$$ecatch \ (ECatch \ p) \ = \ return \ p$$

$$ecatch \ (Other \ op) \ = \ Op \ (fmap \ ecatch \ op)$$

All that needs to change in our example is the type signature, since we now use *Catch* markers before the exception syntax.

$$tripleDecr :: (State \ Int \subset sig, Exc \ () \subset sig, Catch \subset sig) \Rightarrow \\ Prog \ sig \ ()$$

We are now able to change the behaviour by composing *runCatch* and *runState* in different orders:

$$> (run \cdot runCatch \cdot runState \ 2) \ tripleDecr \\ Right \ (1, ())$$

```
> (run · runState 2 · runCatch) tripleDecr
(0, Right ())
```

Here we see that local state behaviour occurs when *runState* is run first, and global state behaviour when it is last.

## 10. Higher-Order Syntax

The previous sections used syntax to carefully mark the beginning and end of syntax blocks that should be handled in some self-contained context. A more direct solution is to model scoping constructs with higher-order syntax, where the syntax carries those syntax blocks directly.

For instance, consider how we might define *HExc*, the higher-order counterpart to the first-order signature *Exc*.

```
data HExc e m a
  = Throw' e
  | ∀x . Catch' (m x) (e → m x) (x → m a)
```

This higher-order signature replaces the scoping handler *catch p h* of *Exc* with scoping syntax *Catch' p h k*. This syntax carries the program contained in *p* directly as an argument, as well as the exception handler *h*. The continuation is *k*, which takes the result from either a successful program *p*, or from the exception handler *h*, depending on whether an exception is thrown.

Note that this higher-order signature refines the single type parameter *cnt* into two parts, *m* and *a*. By having *m* and *a* as two separate type parameters we have tighter control over the type of continuation that is allowed. For instance, in *Catch' p h k* the scoped computation *p* has type *m x* while the continuation *k* turns the result *x* into a computation of type *m a*.

Higher-order signatures are functorial in both type parameters.

In the last parameter  $a$  they must satisfy the ordinary functor laws when  $m$  is a functor. This is the case with  $HExc$ .

**instance** *Functor*  $m \Rightarrow \text{Functor } (HExc\ e\ m)$  **where**

$fmap\ f\ (Throw'\ e) = Throw'\ e$

$fmap\ f\ (Catch'\ p\ h\ k) = Catch'\ p\ h\ (fmap\ f\ \cdot\ k)$

Functoriality in the parameter  $m$  makes such signatures higher-order functors, which are functors in the category of functors and natural transformations. This is captured by the *HFunctor* class.

**type**  $f \dot{\rightarrow} g = \forall x . f\ x \rightarrow g\ x$

**class** *HFunctor*  $h$  **where**

$hmap :: (\text{Functor } f, \text{Functor } g) \Rightarrow (f \dot{\rightarrow} g) \rightarrow (h\ f \dot{\rightarrow} h\ g)$

**instance** *HFunctor*  $(HExc\ e)$  **where**

$hmap\ t\ (Throw'\ x) = Throw'\ x$

$hmap\ t\ (Catch'\ p\ h\ k) = Catch'\ (t\ p)\ (t\ \cdot\ h)\ (t\ \cdot\ k)$

This allows us to transform the type constructor  $m$  with a natural transformation.

## 10.1 Higher-Order Programs

Since higher-order signatures and programs are a generalization of first-order ones, we will redefine all of the infrastructure so that it works in this setting. The type *Prog* now becomes slightly different, since  $sig :: (* \rightarrow *) \rightarrow * \rightarrow *$  is of a different kind. The *Op* constructor is also adjusted accordingly.

**data** *Prog*  $sig\ a$

$= Return\ a$

$| Op\ (sig\ (Prog\ sig)\ a)$

Another important change is in the definition of the monad instance

for *Prog sig a*. In the first-order setting we relied on the functoriality of signatures in the definition of the free monad, where *fmap* was used to substitute syntax into the right part of the resulting syntax

8

tree. In the higher-order setting, *fmap* is inappropriate, since the types do not match. To this end, we introduce the *Syntax* class, whose *emap* function plays the role of *fmap* in the monad instance of *Prog sig a* (we will add more to this typeclass later):

```
class HFunctor sig  $\Rightarrow$  Syntax sig where
  emap :: (Monad m)  $\Rightarrow$  (m a  $\rightarrow$  m b)  $\rightarrow$  (sig m a  $\rightarrow$  sig m b)
  ...
```

The type of *emap* is obtained by refining *cmt* to *m a* in the signature of *fmap*. As an example, consider the definition of *emap* for *HExc*.

```
instance Syntax (HExc e) where
  emap f (Throw' e)    = Throw' e
  emap f (Catch' p h k) = Catch' p h (f  $\cdot$  k)
```

The *Throw'* case is trivial since it does not carry a continuation, and in the *Catch' p h k* case we obviously apply *f* only to the continuation *k*.

Note that *fmap* and *emap* are closely related, and *emap* is subject to similar laws:

$$\text{emap } id = id \tag{1}$$

$$\text{emap } f \cdot \text{emap } g = \text{emap } (f \cdot g) \tag{2}$$

Those versed in category theory will notice that these are the functor laws for the action on arrows, and indeed, we can think of *emap* as a functor from the sub-category obtained through the image of the functor *m*. This gives the following condition:

$$fmap = emap \cdot fmap \quad (3)$$

The *fmap* on the left hand side is from the functor instance of *sig m*, and this should agree with the *fmap* for *m* when extended by *emap*.

We make key use of *emap* in the definition of the free monad *Prog* over higher-order signatures:

```
instance Syntax sig  $\Rightarrow$  Monad (Prog sig) where
  return v = Return v
  Return v >>= prog = prog v
  Op op >>= prog = Op (emap (>>= prog) op)
```

The restricted type of *emap* precisely captures our requirements here, where *m* is *Prog sig*. We leave it as an exercise to show that the monad laws hold.

## 10.2 Higher-Order Infrastructure

The infrastructure that supports higher-order syntax is for the most part an adapted version of what was presented in earlier sections, and its definition is fairly routine. We need only change two things: the *Functor* constraints now become *Syntax* instead, and continuation parameters *cnt* now become *m a*.

As far as the datatypes à la carte machinery is concerned, the only changes are the signatures of the class:

```
class (Syntax sub, Syntax sup)  $\Rightarrow$  sub  $\subset$  sup where
```

$$\begin{aligned} inj &:: sub\ m\ a \rightarrow sup\ m\ a \\ prj &:: sup\ m\ a \rightarrow Maybe\ (sub\ m\ a) \end{aligned}$$

All of the instances need only have their *Functor* constraints turned into *Syntax*, and the bodies remain identical. We can provide patterns for higher-order syntax in just the same way as before:

$$\begin{aligned} \textbf{pattern}\ Throw\ e &\leftarrow (project \rightarrow Just\ (Throw'\ e)) \\ throw &:: (HExc\ e \subset sig) \Rightarrow e \rightarrow Prog\ sig\ a \\ throw\ e &= inject\ (Throw'\ e) \\ \textbf{pattern}\ Catch\ p\ h\ k &\leftarrow (project \rightarrow Just\ (Catch'\ p\ h\ k)) \\ catch &:: (HExc\ e \subset sig) \Rightarrow \\ &\quad Prog\ sig\ a \rightarrow (e \rightarrow Prog\ sig\ a) \rightarrow Prog\ sig\ a \\ catch\ p\ h &= inject\ (Catch'\ p\ h\ return) \end{aligned}$$

```

type HState s = Lift (State s)

pattern Get k ← (project → Just (Lift (Get' k)))
get :: (HState s ⊂ sig) ⇒ Prog sig s
get = inject (Lift (Get' return))

pattern Put s k ← (project → Just (Lift (Put' s k)))
put :: (HState s ⊂ sig) ⇒ s → Prog sig ()
put s = inject (Lift (Put' s (return ())))

```

---

**Figure 2.** Lifted *State* signatures and syntax.

The composition of higher-order syntax is also straight-forward:

```

data (sig1 + sig2) (m :: * → *) a = Inl (sig1 m a) | Inr (sig2 m a)

instance (HFunctor sig1, HFunctor sig2) ⇒
  HFunctor (sig1 + sig2) where
  hmap t (Inl op) = Inl (hmap t op)
  hmap t (Inr op) = Inr (hmap t op)

instance (Syntax sig1, Syntax sig2) ⇒ Syntax (sig1 + sig2) where
  emap f (Inl op) = Inl (emap f op)
  emap f (Inr op) = Inr (emap f op)

```

In addition, we can easily lift our existing first-order signatures to higher-order signatures, by performing the refinement of *cnt* to *m a*. In this lifting, we simply use the first-order *fmap*:

```

newtype (Lift sig) (m :: * → *) a = Lift (sig (m a))

instance Functor sig ⇒ HFunctor (Lift sig) where
  hmap t (Lift op) = Lift (fmap t op)

```



**instance** *Functor sig*  $\Rightarrow$  *Syntax (Lift sig)* **where**  
     $\text{emap } f \text{ (Lift } op) = \text{Lift (fmap } f \text{ } op)$

Providing higher-order signatures and syntax for the first-order counterparts is now simple boilerplate. For instance, running higher-order programs with an empty signature is performed in the same way as first-order programs, except the signature is now lifted:

**type** *HVoid* = *Lift Void*  
 $\text{run} :: \text{Prog } HVoid \text{ } a \rightarrow a$   
 $\text{run (Return } x) = x$

The lifted signatures and syntax for *State* are outlined in Figure 2.

### 10.3 Higher-Order Handlers

With the infrastructure for higher-order syntax in place, we are now in a position to define handlers.

**Handling State** The first higher-order handler we look at is *runState*, which is almost identical to its first-order counterpart. Other than the modification of the signature to use *HState*, most of the definition remains unchanged:

$\text{runState} :: \text{Syntax } sig \Rightarrow$   
     $s \rightarrow \text{Prog (HState } s + sig) \text{ } a \rightarrow \text{Prog } sig \text{ (} s, a)$   
 $\text{runState } s \text{ (Return } a) = \text{return (} s, a)$   
 $\text{runState } s \text{ (Get } k) = \text{runState } s \text{ (} k \text{ } s)$   
 $\text{runState } s \text{ (Put } s' \text{ } k) = \text{runState } s' \text{ } k$

However, the case for *Other op* requires some careful consideration. Using the following definition would be bogus:

$\text{runState } s \text{ (Other } op) = Op \text{ (emap (runState } s) \text{ } op) \quad \text{-- BOGUS!}$

This definition is ill-typed because *emap* only affects the continuation part of *op*, and not any other local computations embedded in *op*. For instance, if *op* were *Catch' p h k*, then *emap (runState s) op*

would result in *Catch p h (runState s · k)*. In other words the state operations in *p* and *h* would go unhandled.

Instead, the generic threading of a handler *hdl* through *Catch' p h k* should result in *Catch' p' h' k'*. When no exception is thrown in *p*, then  $p' \gg=k'$  should be equivalent to running *hdl* on  $p \gg=k$ . This means that it should be possible to suspend handlers and resume them from an intermediate point, for example between *p* and *k*.

**Distributive laws** Since a handler such as *runState* may be stateful, we need to capture its intermediate state in a (functorial) context *c* when it suspends and make it available when resuming. This motivates us to consider handlers of the type:

$$hdl :: \forall x . c (m x) \rightarrow n (c x)$$

This defines a higher-order handler as a natural transformation that takes a state-annotated computation in one monad *m* into a computation in another monad *n* whose value is in a stateful context. If we uncurry *runState*, its type matches this pattern with  $m = \text{Prog } (H\text{State } s + \text{sig})$ ,  $n = \text{Prog } \text{sig}$  and  $c = (, ) s$ .

By further imposing the following conditions, a handler becomes a well-behaved distributive law:

$$hdl \cdot \text{fmap return} = \text{return} \tag{4}$$

$$hdl \cdot \text{fmap join} = \text{join} \cdot \text{fmap hdl} \cdot hdl \tag{5}$$

These laws arise naturally as the coherence conditions that govern the interaction between *hdl* and monadic computations.

The first law expresses that the handler preserves a pure computation without modifying its state. The second law expresses that it makes no difference whether a composite computation is transformed before or after composition, and makes use of *join*, which is interdefinable with ( $\gg\equiv$ ):

$$\text{join } mx = mx \gg\equiv id \qquad mx \gg\equiv f = \text{join } (\text{fmap } f \text{ } mx)$$

While the second distributivity law is better expressed in terms of *join*, an operational understanding might be more easily obtained when we rewrite it using these equivalences and name values:

$$\text{hdl } (\text{fmap } (\gg\equiv k) \text{ } sm) = \text{hdl } sm \gg\equiv \text{hdl } \cdot \text{fmap } k$$

**Distributing Handlers** In order to make use of such distributive laws, each fragment of syntax needs to define its own scheme for distributing such a handler through its structure. Therefore, we augment the *Syntax* typeclass with a *handle* method in addition to *emap*:

```
class HFunctor sig  $\Rightarrow$  Syntax sig where
  emap :: (Monad m)  $\Rightarrow$  (m a  $\rightarrow$  m b)  $\rightarrow$  (sig m a  $\rightarrow$  sig m b)
  handle :: (Monad m, Monad n, Functor c)  $\Rightarrow$ 
    c ()  $\rightarrow$  ( $\forall x$  . c (m x)  $\rightarrow$  n (c x))  $\rightarrow$  (sig m a  $\rightarrow$  sig n (c a))
```

The *handle* method generically threads a handler through a higher-order signature. In addition to the handler, the method also takes the initial context of *c*, which is represented by *c ()*.

Since we extended the definition of *Syntax*, we must also define a *handle* instance for the higher-order infrastructure.

```

instance (Syntax sig1, Syntax sig2) ⇒ Syntax (sig1 + sig2) where
  ...
  handle c hdl (Inl op) = Inl (handle c hdl op)
  handle c hdl (Inr op) = Inr (handle c hdl op)
instance Functor sig ⇒ Syntax (Lift sig) where
  ...
  handle c hdl (Lift op) =
    Lift (fmap (λp → hdl (fmap (const p) c)) op)

```

With this additional machinery in place, we can finally complete the definition of *runState*:

$$\text{runState } s \text{ (Other } op) = Op \text{ (handle } (s, ()) \text{ (uncurry runState) } op)$$

Here we see that the  $fmap (\text{runState } s) \text{ } op$  in the original definition has been replaced with  $\text{handle } (s, ()) \text{ (uncurry runState) } op$ , where  $(,) s$  is the context functor.

In order to distribute this handler through the  $HExc$  syntax, we need to provide the  $HExc$  definition of  $\text{handle}$ :

**instance** *Syntax* ( $HExc \text{ } e$ ) **where**

...

$$\begin{aligned} \text{handle } c \text{ hdl } (Throw' \text{ } x) &= Throw' \text{ } x \\ \text{handle } c \text{ hdl } (Catch' \text{ } p \text{ } h \text{ } k) &= \\ &\quad Catch' \text{ } (\text{hdl } (fmap \text{ } (const \text{ } p) \text{ } c)) \\ &\quad (\lambda e \rightarrow \text{hdl } (fmap \text{ } (const \text{ } (h \text{ } e)) \text{ } c)) \\ &\quad (\text{hdl } \cdot fmap \text{ } k) \end{aligned}$$

The  $Throw'$  case is trivial, but  $Catch'$  is more involved. We derive its implementation as follows. We assume that the result of  $\text{handle } c \text{ hdl } (Catch' \text{ } p \text{ } h \text{ } k)$  has the form  $Catch' \text{ } p' \text{ } h' \text{ } k'$ . In the absence of exceptions in  $p$ , we intuitively require that

$$p' \gg= k' = \text{hdl } (fmap \text{ } (\gg= k) \text{ } (fmap \text{ } (const \text{ } p) \text{ } c))$$

In other words, executing a successful computation  $p'$  followed by  $k'$  should be the result of handling the computation that starts in a context  $c$ , where  $p$  is executed, followed by  $k$ .

Based on the second handler law, the latter is equivalent to

$$\text{hdl } (fmap \text{ } (const \text{ } p) \text{ } c) \gg= \text{hdl } \cdot fmap \text{ } k$$

which gives us the following solutions for  $p'$  and  $k'$  in terms of  $p$  and  $k$  respectively:

$$p' = \text{hdl } (fmap \text{ } (const \text{ } p) \text{ } c) \qquad k' = \text{hdl } \cdot fmap \text{ } k$$

Similarly, in the case where  $p = \text{throw } e$ , we require that

$$h' e \ggg k' = \text{hdl } (\ggg k) (\text{fmap } (\text{const } (h' e)) c)$$

In other words, handling an exception  $e$  with  $h'$  and proceeding with  $k'$  should be the same as starting in the context  $c$  where  $h e$  is executed, followed by  $k$ . Applying the second handler law again gives us a solution:

$$h' = \lambda e \rightarrow \text{hdl } (\text{fmap } (\text{const } (h e)) c)$$

**Handling Exceptions** The higher-order handler for exceptions is *runExc*, and its definition is unsurprisingly more complex than *runState*, since we must take care of the intermediate state that is to be woven through higher-order syntax. The clauses for *Return* and *Throw* remain identical, and only *Catch* and *Other* are of interest.

*runExc* :: Syntax sig  $\Rightarrow$

*Prog* (HExc  $e + \text{sig}$ )  $a \rightarrow \text{Prog sig (Either } e a)$

*runExc* (*Return*  $x$ ) = *return* (*Right*  $x$ )

*runExc* (*Throw*  $x$ ) = *return* (*Left*  $x$ )

*runExc* (*Catch*  $p h k$ ) =

**do**  $r \leftarrow \text{runExc } p$

**case**  $r$  **of** *Left*  $e \rightarrow \text{do } r \leftarrow \text{runExc } (h e)$

**case**  $r$  **of**

*Left*  $e \rightarrow \text{return } (\text{Left } e)$

*Right*  $a \rightarrow \text{runExc } (k a)$

*Right*  $a \rightarrow \text{runExc } (k a)$

*runExc* (*Other*  $op$ ) = *Op* (*handle* (*Right*  $()$ ) *hdl op*) **where**

*hdl* :: Syntax sig  $\Rightarrow (\forall x . \text{Either } e (\text{Prog } (\text{HExc } e + \text{sig}) x) \rightarrow$

*Prog sig (Either } e x))*

*hdl* = *either* (*return* · *Left*) *runExc*

The case for *Catch p h k* attempts to execute *p*. If an exception *e* is thrown, then this is used to execute the exception handler *h*. If that fails again, then this is an uncaught exception that propagates its

10

way out. In all other cases, some valid computation *a* is returned, which is fed into *runExc (k a)*, since *k a* may itself throw exceptions.

The *Other* case starts computations with a successful computation *Right ()*. The distributive handler simply propagates errors when they are encountered with *return · Left*, and otherwise recursively applies *runExc*.

Finally, recreating the example of the previous section requires very little work indeed. For instance, here is the definition of *tripleDecr*:

$$\begin{aligned} tripleDecr &:: (HState Int \subset sig, HExc () \subset sig) \Rightarrow Prog sig () \\ tripleDecr &= decr \gg (catch (decr \gg decr) return) \end{aligned}$$

The only change is in the *types* of the functions, since we have moved to a higher-order setting!

## 11. Multi-Threading

Cooperative multi-threading allows a thread to suspend its computation with a *yield* operation: this relinquishes control to the scheduler, which may decide to run a different thread. We assume the existence of an initial master thread, and new child threads are created by calling *fork*.

The examples of scoped effects we have seen so far—pruning

non-deterministic computations, and exception handling—have been solved both by clever use of first-order syntax tagging, and also by higher-order syntax. We might, therefore, expect both techniques to be equally expressive. In this section we put that expectation to rest and show that cooperative multi-threading can only be solved using the higher-order approach.

## 11.1 Signature

The multi-threading effect *Thread* concerns two operations: *fork d* spawns a new thread *d*, and *yield* relinquishes control. While *yield* is a plain algebraic operation, *fork* is clearly a scoping construct that delimits the new thread.

It would be wrong to capture the signature as the following first-order syntax.

```
data Thread cnt  -- BOGUS!
    = Yield' cnt
    | Fork' cnt cnt
```

Here *Fork' p q* would represent a computation that spawns a new thread *p*, while the master thread continues with *q*. The problem is that, in the first-order framework, we have that

$$Op (Fork' p q) \ggg k = Op (Fork' (p \ggg k) (q \ggg k))$$

This is clearly not the desired semantics, since forking would result in the continuation *k* being run in both the parent thread and its child. Instead we want the following, where the continuation is only applied to the remainder of the parent *q*.

$$Op (Fork' p q) \ggg k = Op (Fork' p (q \ggg k))$$



We should distinguish between the subcomputation for the child thread and the one for the continuation of the parent. First-order syntax does not have this capability, only higher-order syntax does:

**data** *Thread* *m a*

*= Yield'* (*m a*)

|  $\forall x . \text{Fork}' (m\ x) (m\ a)$

**pattern** *Yield* *p*  $\leftarrow (\text{project} \rightarrow \text{Just } (\text{Yield}'\ p))$

*yield* :: (*Thread*  $\subset$  *sig*)  $\Rightarrow$  *Prog sig* ()

*yield* = *inject* (*Yield'* (*return* ()))

**pattern** *Fork* *p q*  $\leftarrow (\text{project} \rightarrow \text{Just } (\text{Fork}'\ p\ q))$

*fork* :: (*Thread*  $\subset$  *sig*)  $\Rightarrow$  *Prog sig a*  $\rightarrow$  *Prog sig* ()

*fork* *d* = *inject* (*Fork'* *d* (*return* ()))

The *Syntax* instance shown below distinguishes between the two subcomputations, where *emap* marks the continuation and ensures the desired semantics for  $(\gg=)$ .

**instance** *Syntax Thread* **where**

$$\mathit{emap} f \ (\mathit{Yield}'\ p) \quad = \quad \mathit{Yield}'\ (f\ p)$$

$$\mathit{emap} f \ (\mathit{Fork}' \ d \ p) = \quad \mathit{Fork}' \ d \ (f \ p)$$

$$\text{handle } c \text{ hdl } (\text{Yield}' p) = \text{Yield}' (\text{hdl } (\text{fmap } (\text{const } p) c))$$

$$\text{handle } c \text{ hdl } (\text{Fork}' \ d \ p) = \text{Fork}' \ (\text{hdl } (\text{fmap } (\text{const } d) \ c)) \ (\text{hdl } (\text{fmap } (\text{const } p) \ c))$$

Note that the result type of the new thread is locally quantified:

$$Fork' :: \forall x . m\ x \rightarrow m\ a \rightarrow Thread\ m\ a$$

This is in line with the notion that, in *Fork'*  $dp \ggg k$ , the continuation  $k$  does not interact with the child's result: there is no direct communication between the child and the master thread, and no need to constrain its return type.

We call a thread with an existentially quantified result type a *daemon*, in contrast with the master thread of the program.

$$\mathbf{data} \text{ Daemon sig} = \forall x . \text{Daemon} (\text{Prog} (\text{Thread} + \text{sig}) x)$$

## 11.2 Handler

We adopt the semantics that a thread suspends at every *fork* and *yield* in favour of running another thread. Hence, a thread can be in one of three different states, two of which are named after the corresponding syntax:

```
data SThread sig r
```

$$= SYield \ (Prog \ (Thread + sig) \ r)$$

$$SFork \quad (Daemon \ sig) \ (Prog \ (Thread + sig) \ r)$$

| *SActive r*

**instance** *Syntax sig*  $\Rightarrow$  *Functor (SThread sig)* **where**

*fmap f (SActive x) = SActive (f x)*

*fmap f (SYield p) = SYield (liftM f p)*

*fmap f (SFork d p) = SFork d (liftM f p)*

The default state is *SActive*, which denotes an ongoing thread. The *SFork d p* state denotes a thread that has suspended at a *fork* that spawns a daemon *d* and should continue with *p*. Similarly, *SYield p* denotes a thread that has suspended at a *yield* and should continue with *p*. The similarity between *Thread* and *SThread* is no coincidence: *SThread* is very nearly *Prog (Thread + sig) a*, except that the latter has an additional constructor *Other* for syntax.

The *runThread* handler runs a thread and returns its resulting state in the obvious way.

*runThread :: Syntax sig*  $\Rightarrow$

*Prog (Thread + sig) x*  $\rightarrow$  *Prog sig (SThread sig x)*

*runThread (Return x) = return (SActive x)*

*runThread (Yield q) = return (SYield q)*

*runThread (Fork d q) = return (SFork (Daemon d) q)*

*runThread (Other op) = Op (handle (SActive ()) thread op)*

The helper function *thread* decides how to continue from an intermediate state. It calls *runThread* for an active thread, and extends the continuation for both kinds of suspended threads.

*thread :: Syntax sig*  $\Rightarrow \forall x . (SThread sig (Prog (Thread + sig) x)$   
 $\rightarrow Prog sig (SThread sig x))$

*thread (SActive p) = runThread p*

*thread (SYield p) = return (SYield (join p))*

*thread (SFork d p) = return (SFork d (join p))*

Finally, the top-level *schedule* handler runs the master thread and daemons in round-robin fashion. It switches from one thread to another at every *fork* and *yield*.

11

```

schedule :: Syntax sig ⇒ Prog (Thread + sig) a → Prog sig a
schedule p = master p [] where
  master p ds =
    do r ← runThread p
    case r of
      SActive x → return x
      SYield p → daemons ds [] p
      SFork d p → daemons (d : ds) [] p
  daemons [] ds' p = master p (reverse ds')
  daemons (Daemon q : ds) ds' p =
    do r ← runThread q
    case r of
      SActive _ → daemons ds ds' p
      SYield q' → daemons ds (Daemon q' : ds') p
      SFork d' q' → daemons (d' : ds) (Daemon q' : ds') p

```

The *schedule* function adopts the termination condition of the Go language:<sup>2</sup> the whole program ends when the master thread ends; any unfinished daemons are discarded.

### 11.3 Threading in Action

In order to show an example of multi-threading behaviour, we will introduce syntax for communicating to the outside world with IO.

The syntax *Out' x* expresses that *x* should be written out.

**data** *Out cnt = Out' String cnt*

**type** *HOut = Lift Out*

**pattern** *Out s p*  $\leftarrow$  (*project*  $\rightarrow$  *Just (Lift (Out' s p))*)

*out* :: (*HOut*  $\subset$  *sig*)  $\Rightarrow$  *String*  $\rightarrow$  *Prog sig* ()

*out s = inject (Lift (Out' s (return ())))*

The *io* handler turns the syntax into semantics by invoking the appropriate OS system call, which we model by *putStr*.

*io* :: *Prog HOut a*  $\rightarrow$  *IO a*

*io (Return x)* = *return x*

*io (Out s p)* = **do** *putStr s*; *io p*

With this in place, the following program combines the multi-threading effect with state and output.

*prog* :: (*Thread*  $\subset$  *sig*, *HState Int*  $\subset$  *sig*, *HOut*  $\subset$  *sig*)  $\Rightarrow$  *Prog sig* ()

*prog* = **do** *logIncr "master"*

*fork (logIncr "daemon" >> logIncr "daemon")*

*logIncr "master"*

The master thread increments the state twice, but spawns a new thread in between that increments the state twice too.

*log* :: (*HState Int*  $\subset$  *sig*, *HOut*  $\subset$  *sig*)  $\Rightarrow$  *String*  $\rightarrow$  *Prog sig* ()

*log x* = **do** (*n* :: *Int*)  $\leftarrow$  *get*; *out (x ++ " : " ++ show n ++ " ; ")*

*logIncr* :: (*HState Int*  $\subset$  *sig*, *HOut*  $\subset$  *sig*)  $\Rightarrow$  *String*  $\rightarrow$  *Prog sig* ()

*logIncr x* = *log x* >> *incr*

The *logIncr* function outputs the state before incrementing it.

With our for multi-threading we can obtain two different semantics for *prog*: one where the state is shared among all threads, and one where it is local to each thread. By running *runThread* before *runState* we share the state between the master and daemon:

```
> (io · runState (0::Int) · schedule) prog  
master: 0; daemon: 1; daemon: 2; master: 3;  
(4,())
```

The other way around, *fork* creates a local copy of the state for the daemon and updates are not shared:

---

<sup>2</sup><http://golang.org/>

```
> (io · schedule · runState (0 :: Int)) prog
master: 0; daemon: 1; daemon: 2; master: 1;
(2,())
```

We have successfully shown how higher-order syntax allows effects to be scoped, which has resulted in interesting compositional semantics through the simple reordering of handlers.

## 12. Related Work

### 12.1 Effect Handlers

Plotkin and Power were the first to explore effect operations [11], and gave an algebraic account of effects [12] and their combination [4]. Subsequently, Plotkin and Pretnar [13] have added the concept of handlers to deal with exceptions. This has led to many implementations.

**Languages** Based on this idea, two entirely new programming languages have been created from the ground up around algebraic effect handlers.

- Eff [1] is an ML-variant that does not track effect signatures in its static type system. Hence, its type system does not rule out higher-order syntax. For instance, Bauer and Pretnar show how to implement the multi-threading example in Eff, but can only get the global state interpretation.
- Frank [10] does track effect signatures in its static type system, but does not allow higher-order syntax.

**Libraries** More recently, three proposals show how to implement algebraic effect handlers on top of existing functional programming languages:

- Brady [2] provides an effect handlers approach in the Idris lan-

guage. The approach supports only one special built-in syntactic scoping construct, *catch*, but does not support additional higher-order syntax.

- Kammar et al. [7] present several different implementations in Haskell, OCaml, SML and Racket. These are based on different implementation techniques: the free monad and a continuation-based approach are considered in Haskell, and delimited continuations for the other languages. Scoping syntax is not covered. How delimited continuations can be used to implement higher-order syntax requires further investigation.
- Kiselyov et al. [8] provide a Haskell implementation in terms of the free monad, in combination with two optimizations: 1) the codensity transformer improves the performance of ( $\gg=$ ), and 2) their *Dynamic*-based open unions have a better time complexity than nested co-products. We believe that both of these optimizations can be applied to our first-order and higher-order scoped syntax. They do not cover scoping syntax.

As far as we are aware Kiselyov et al. were the first to provide a handler for *Cut*,<sup>3</sup> inspired by Hinze’s monad transformer [3]. However, they do not discuss the scoping problem.

## 12.2 Monad Transformers

The issue of scoping operations already arises in the more general setting of monad transformers [9] where different effects can be composed in different ways to obtain different semantics. The *lift* operation is used to combine operations from different transformers in the same program. While lifting algebraic operations is typically easy, lifting scoping operations is typically not. This problem is addressed by Jaskelioff and Moggi [6] for a class of *functorial* operations and available in the Monatron library [5].

---



<sup>3</sup> Our variant in Section 5 simplifies theirs from  $n$ -way to 2-way choice and omits the codensity optimization.

## 13. Conclusion

We have shown that using effect handlers for scoping fixes the interaction between effects. Our main message is that, in order to regain control over the semantics of interaction, syntax should determine scope.

We provide two approaches for scoping syntax, scope markers and higher-order syntax, each with their merits. Scope markers play nicely with all existing effect handler frameworks and can piggy-back on existing optimizations and convenience infrastructure (e.g., Template Haskell macros). In contrast, it is an open question how higher-order syntax can be implemented on top of delimited continuations, which is the basis for effect handlers in strict languages. The advantages of higher-order syntax are that it is strictly more expressive and that is a more natural way to denote scoping; the markers run the risk of being unbalanced.

For future work we believe that it should be possible to generically lift first-order handlers to the higher-order setting provided that they are expressed in terms of algebras.

## Acknowledgments

This work has been funded by EPSRC grant number EP/J010995/1 on Unifying Theories of Generic Programming, and by the Flemish

## References

- [1] A. Bauer and M. Pretnar. Programming with algebraic effects and handlers, 2012.
- [2] E. Brady. Programming and reasoning with algebraic effects and dependent types. In *Proceedings of the 18th ACM SIGPLAN International Conference on Functional programming*, pages 133–144. ACM, 2013.
- [3] R. Hinze. Deriving backtracking monad transformers. In *Proceedings of the Fifth ACM SIGPLAN International Conference on Functional Programming*, pages 186–197. ACM, 2000. ISBN 1-58113-202-6. .
- [4] M. Hyland, G. D. Plotkin, and J. Power. Combining effects: Sum and tensor. *Theor. Comput. Sci.*, 357(1-3):70–99, 2006.
- [5] M. Jaskelioff. Monatron: An extensible monad transformer library. In S.-B. Scholz and O. Chitil, editors, *Implementation and Application of Functional Languages - Revised Selected Papers*, volume 5836 of *Lecture Notes in Computer Science*, pages 233–248. Springer, 2008.
- [6] M. Jaskelioff and E. Moggi. Monad transformers as monoid transformers. *Theor. Comput. Sci.*, 411(51-52):4441–4466, Dec. 2010.
- [7] O. Kammar, S. Lindley, and N. Oury. Handlers in action. In *Proceedings of the 18th ACM SIGPLAN International Conference on Functional programming*, pages 145–158. ACM, 2013.
- [8] O. Kiselyov, A. Sabry, and C. Swords. Extensible effects: an alternative to monad transformers. In *Proceedings of the 2013 ACM SIGPLAN symposium on Haskell*, Haskell ’13, pages 59–70. ACM, 2013.
- [9] S. Liang, P. Hudak, and M. Jones. Monad transformers and modular interpreters. In *POPL’95*, 1995.
- [10] C. McBride. The Frank manual, May 2012. <https://personal.cis.strath.ac.uk/conor.mcbride/pub/Frank/TFM.pdf>.

- [11] G. D. Plotkin and J. Power. Notions of computation determine monads. In M. Nielsen and U. Engberg, editors, *Foundations of Software Science and Computation Structures*, volume 2303 of *Lecture Notes in Computer Science*, pages 342–356. Springer, 2002.
- [12] G. D. Plotkin and J. Power. Algebraic operations and generic effects. *Applied Categorical Structures*, 11(1):69–94, 2003.
- [13] G. D. Plotkin and M. Pretnar. Handlers of algebraic effects. In *ESOP*, volume 5502 of *Lecture Notes in Computer Science*, pages 80–94. Springer, 2009.
- [14] W. Swierstra. Data types à la carte. *J. Funct. Program.*, 18(4):423–436, 2008.